
Глава 1
Введение в MetaPost

Каждая иллюстрация — это целая история.

Люди делятся на тех, кто умеет рисовать, и тех, кто не умеет. Причём вторых
большинство. В этом нет ничего плохого или хорошего — такова жизнь. Хороший
иллюстратор — редкость.

Компьютерные технологии дают возможность создавать высококачественные до-
кументы особенно не концентрируясь на проблеме оформления. Всю черновую ра-
боту делает за вас компьютер. Например, система LATEX фактически заменяет собой
электронную типографию, но, к сожалению, создание иллюстраций выносится за
рамки этого процесса. Умение иллюстрировать свой текст является необходимым
навыком для тех, кто хочет делать свои книги самостоятельно от начала и до конца.
Часто бывает что основной смысл несут в себе именно рисунки.

Создание иллюстраций это длительный и тяжёлый процесс, но если целью стоит
совершенство вашей книги, то потраченное время того стоит. Обучение в изостудии
в младших классах подняло мой уровень рисования с «никуда не годного» до «тер-
пимого, если отойти на сто метров». Как художник я совершенно безнадёжен, но
иллюстрации мне приходится делать довольно часто. В основном это не сложные
картинки к задачам по физике. Когда мне говорят, что мои рисунки к задачам вы-
зывают чувство зависти — мне с одной стороны приятно, но с другой стороны очень
тяжело объяснить как же я этого достиг. Цель этого текста — популяризовать со-
здание иллюстраций с помощью MetaPost.

Желание контролировать всё в процессе создания книги в случае Дональда Э.
Кнута (Donald E. Knuth) привело к созданию программ TEX и METAFONT.

Изначально METAFONT предназначался для создания шрифтов и результатом его
работы был растр с изображением шрифта. Позже аспирант Д.Э.Кнута Джон Хоб-
би (John Hobby) модифицировал METAFONT таким образом, что результатом работы
программы стала картинка в формате EPS (Encusulated PostScript).

MetaPost — это программа для тех, кто может объяснить компьютеру что он хо-
чет. MetaPost чрезвычайна полезен для случаев, когда картинку проще описать
логически, нежели образно. Даже если Вы не умеете рисовать, результат может

1 Введение в MetaPost

получиться вполне приличный, потому что «виртуальный карандаш» под управле-
нием компьютера дрожать не будет.

MetaPost проектировался как простая программа, которую можно относитель-
но быстро настроить под свои нужды. Он вполне обозрим и компактен. Многие
«правильные» программы, работающие с векторной графикой, такие как gnuplot

и xfig, умеют экспортировать в MetaPost.

1.1. Здравствуй, Мир!

Когда изучается новая программная технология первое что надо сделать, это
сказать: «Здравствуй, Мир!». Что и сделаем, правда, пока на английском. Для этого
следует создать файл helloworld.mp со следующим содержанием:

% Для просмотра

prologues := 1 ;
% Простой Hello , World!

beginfig ( 1 ) ;
label ( "Hel lo , ␣World ! ␣␣␣␣␣" , ( 0 , 0 ) ) ;

endfig ;
% <<конец обработки >> - необходимо вставить в конце файла

end .

Далее скомпилируем этот код:

> mpost he l l owor ld .mp

В результате полученный файл helloworld.1 можно посмотреть с помощью любой
программой понимающей PostScript, например, с помощью gv:

Hello, World!
Рис. 1.1. Первый Hello World на MetaPost

Разнообразим пример. Создадим файл linuxformat.mp со следующим содержи-
мым:

% А теперь мы его завращали и раскрасили

beginfig ( 1 ) ;
for alpha :=90 step −3 until 0 :

label (btex Linux Format etex
scaled (5∗(1− alpha /100)) rotated alpha , ( 0 , 0 ) )

withcolor (max(1−alpha /45 ,0)∗ red+
min( alpha /45,2−alpha /45)∗green+

2



1 Введение в MetaPost

max( alpha /45−1 ,0)∗blue ) ;
endfor ;

endfig ;
end .

Скомпилируем код и результат преобразуем в pdf:

> mpost l inuxformat .mp
> pstopdf l inuxformat . 1

Полученный в результате этой операции файл linuxformat-1.pdf можно посмот-
реть, например, с помощью программы acroread или xpdf.

Рис. 1.2. «Продвинутый» Hello World на MetaPost

Всё, что находится между btex и etex обрабатываются внешней программой. По
умолчанию это TEX. Чтобы воспользоваться возможностями LATEX надо в самом
начале linuxformat.mp строчки вида:

verbat imtex \documentclass{ a r t i c l e }
\ beg in {document}
etex ;

Эти строчки указывают mpost, что при обработке любых текстовых вставок им
должны предшествовать команды \documentclass{article} и \begin{document}.

Добавим в linuxformat.mp ещё один рисунок. Для этого скопируем уже име-
ющийся код первого рисунка, вставим его до заключительной команды «end.» и
немного подправим первые три строчки:

% Математика вместо Hello , World!

beginfig ( 2 ) ;
for alpha :=90 step −9 until 0 :

label (btex \( f (x)=\ f r a c {1}{\ sqrt {2\ p i }\ ,\ sigma}
\ i n t \ l imi t s_{−\ i n f t y }^{\ i n f t y }

e^{−\ f r a c {x^2}{2\sigma^2}}dx\) etex
. . .

3



1 Введение в MetaPost

Создание новых картинок на базе уже имеющегося кода в MetaPost — обычное де-
ло. Со временем накапливается своя библиотечка примитивов, что сильно облегчает
создание новых иллюстраций. Кажущаяся сложность выражений компенсируется
исключительной гибкостью языка, позволяющая повторно использовать набранный
код.

Скомпилируем код ещё раз:

> mpost −tex=l a t ex l inuxformat .mp
> pstopdf l inuxformat . 2

Таким образом можно отобразить любую сколь угодно сложную математическую
конструкцию:

Рис. 1.3. Математическая вставка

Вместо того чтобы запускать mpost с опцией -tex=latex, можно установить пе-
ременную окружения TEX равную latex. В bash это можно сделать так:

> export TEX=la t ex

Как видно из предыдущего примера, MetaPost и LATEX умеют неплохо работать
вместе. Усовершенствуем немного linuxformat.mp. Заменим первые три строчки в
файле на следующий код:

verbat imtex \ input{ preheader }
\ beg in {document}
etex ;

и создадим файл preheader.tex с примерно следующим содержанием:

\ documentclass [ 12 pt ] { a r t i c l e }
% простейшая кириллизация

\usepackage [ ko i8−r ] { inputenc }
\usepackage [ eng l i sh , r u s s i an ] { babel }
\ usepackage { i n d e n t f i r s t }%first paragraph indent

\usepackage { graphicx }

4



1 Введение в MetaPost

Это стандартный заголовок для файлов LATEX. Если Вы набираете свои тексты в
LATEX, то можете поместить в preheader.tex всю свою преамбулу и включить её с
помощью команды \input. Это позволит создавать текстовые вставки в картинках
MetaPost тем же шрифтом, что и в обычном тексте, а также позволит пользоваться
любыми предопределёнными вами командами.

Следует обратить внимание, что чем больше будет разрастаться заголовок, тем
дольше будет обрабатываться META-код. Для увеличения скорости лучше всего вос-
пользоваться классом minimal, вместо article и убрать все ненужные для рисова-
ния картинок пакеты.

Добавим третью картинку в linuxformat.mp, чуть изменив первоначальный код:

verbat imtex \ input{ preheader }
\ beg in {document}
etex ;

. . .

% изменим направление вращения

beginfig ( 3 ) ;
for alpha :=−90 step 3 until 0 :

label (btex LinuxFormat в России etex
scaled (5∗(1+alpha /100)) rotated alpha , ( 0 , 0 ) )

withcolor (max(1+alpha /45 ,0)∗ red+
min(−alpha /45 ,2+alpha /45)∗green+
max(−alpha /45−1 ,0)∗blue ) ;

endfor ;
endfig ;
end .

Теперь мы можем говорить по русски.

Рис. 1.4. Говорим по русски

1 Введение в MetaPost

META-картинка

mpost

PostScript без шрифтов

LATEX-«обёртка»

pdf

META-картинка

latex+dvips

eps

Рис. 1.5. MetaPost-конвейер

1.2. MetaPost-конвейер

Чуть подробнее остановимся на том что же происходит. На вход программы
metapost подаётся «META-картинка». «META-картинка» — это текстовый mp-файл с
инструкциями на языке META. В одном mp-файле можно хранить несколько опи-
саний картинок. При компиляции с помощью mpost создаются файлы с тем же
именем, что и у исходного файла, но с расширениями в виде чисел, которые ука-
зываются в декларации beginfig. Результирующие файлы сразу можно вставлять
в LATEX-тексты с помощью обычного \includegraphics. Для этого достаточно в
заголовок tex-файла добавить команду из LATEX-пакета graphicx:

\DeclareGraphicsRule {∗}{ eps }{∗}{}

От «правильных» eps-файлов они отличаются только тем, что в них не «внедрены»
шрифты, поэтому просмотреть их без дополнительной обработки не удастся.

Шрифты можно внедрить посредством программ latex и dvips с результатом в
виде eps-файла или скрипта mptopdf с результатом в виде pdf-файла. Эти картинки
уже можно использовать независимо любой программой, которая поддерживает эти
векторные форматы.

1.3. Среда разработки

Удобная среда разработки понятие весьма относительное. Удобство зависит ис-
ключительно от Ваших предпочтений. Здесь будет представлена среда удобная

для меня. Подробности настройки и установки упоминаемых здесь пакетов выходят
за рамки этого теста.

В качестве базовой операционной системы я использую GNU/Linux. Это позволя-
ет мне использовать для своей работы стандартные средства обработки текстовых
данных, которые поставляются с этой системой. Например, ниже будет упомянута
утилита make, которая позволяет автоматизировать ряд стандартных действий.

В качестве базового дистрибутива LATEX я использую дистрибутив TEXLive. На
сегодня этот дистрибутив является наиболее полным из дистрибутивов LATEX. Уста-
новочный образ для CD можно взять на любом из CTAN-архивов (Comprehensive TEX
Archive Network).

В качестве базового редактора я использую редактор emacs. Если Вы работаете
с текстами в LATEX-формате, то при настройки этого редактора следует включить
пакеты auctex1 и reftex2. emacs имеет простейшую встроенную поддержку для
редактирования файлов в формате MetaPost.

Для автоматизации создания eps-картинок я использую Makefile примерно сле-
дующего вида:

# временный файл

tmp_fi le := tmp_fi le
# программы

LATEX := la t ex
MPOST := mpost −tex=l a t ex
DVIPS := dvips

a l l :
@echo "run : ␣make␣mpf i l e . n . eps "

%.eps : % preheader . tex
@echo "\ input { preheader }">$ ( tmp_fi le ) . tex
@echo "\DeclareGraphicsRule {∗}{ eps }{∗}{}">>\

$ ( tmp_fi le ) . tex
@echo "\ n o f i l e s ">>$ ( tmp_fi le ) . tex
@echo "\ beg in {document}">> $( tmp_fi le ) . tex
@echo "\ t h i s p a g e s t y l e {empty}">> $ ( tmp_fi le ) . tex
@echo "\ i n c l udeg r aph i c s {$ ( basename␣$@)} ">>\

$ ( tmp_fi le ) . tex
@echo "\end{document}">> $( tmp_fi le ) . tex
@$(LATEX) $ ( tmp_fi le )
@$(DVIPS) −E −o $@ $ ( tmp_fi le )
@rm $ ( tmp_fi le ) . ∗

c l ean :

1Продвинутая система для работы с проектами LATEX.
2Автоматизация работы со ссылками.

@rm −f mpx∗ ∗~ ∗ . l og ∗ .mpx
@rm −f $ ( tmp_fi le ) . ∗

# Зависимости для mpost -картинок.

# По одной для каждого числа из beginfig

%.1: %.mp preheader . tex preheader .mp
$ (MPOST) $<

. . .
%.64: %.mp preheader . tex preheader .mp

$ (MPOST) $<

Чтобы на выходе получить готовую eps-картинку с уже «внедрёнными» шриф-
тами, которую можно вставить уже куда угодно достаточно выполнить следующую
команду:

> make <имя mp−файлы>.<номер картинки>. eps

Обычно, mp-файлам даются короткие имена.
Если в качестве результирующего формата картинок вас интересует pdf-формат,

то в Makefile можно добавить что-то вроде:

MPTOPDF := mptopdf
MV := mv

%. pdf : % preheader . tex
@$(MPTOPDF) $<
@$(MV) ‘ echo $< | sed −e \

" s /\ . \( [0−9 ] \+\)$$/−\1 . pdf /" ‘ $<. pdf

1.4. Чуть-чуть о META

В качестве базового языка, инструкции которого подаются на вход программы
MetaPost используется язык META.

Язык META поддерживает следующие типы данных:

boolean объект булева типа, то есть либо true, либо false,

numeric обычное число (по умолчанию, если переменная не описана то она счита-
ется numeric),

pair точка — пара чисел (x,y) в случае декартовых координат или R∗dir α в случае
полярных координат,

pen перо — то, чем компьютер рисует (в подавляющем большинстве случаев ис-
пользуется круглое перо pencircle),

color цвет — тройка чисел (r,g,b),

path путь — совокупность точек с описанием соединения между ними,

picture картинка — совокупность путей и точек,

string строка —ASCII строка,

transform преобразования— линейные преобразования, которые можно применять
к объектам типа pair, pen, path и picture.

Имена переменных в META могут состоять из нескольких лексем. Лексемы могут
быть либо буквенными, либо числовыми. Например, переменная x1l состоит из трёх
лексем. Её можно переписать более понятным способом x[1].l, то есть числовая
лексема по сути указывает на номер элемента в массиве, а следующая за ней буква
уточняет элемент структуры. Возможность упускать «[].» в написании имён пе-
ременных упрощает в некоторых случаях восприятие кода (x1l — это x-координата
границы линии слева по направлению движения для первой точки пути z[]) и
сокращает объём программы. В замен, если Вам нужны просто переменные без
подобных особенностей, то Вам придётся ограничиться только буквенными комби-
нациями.

Все переменные необходимо объявлять перед использованием. Исключением яв-
ляются переменные типа numeric. Массивы объявляются и используются следую-
щим образом:

pair w [ ] ;
w1 :=(10 , 5 ) ;
w[2 ]=w [ 1 ] ;

Взаимодействие переменных, чисел и операторов вполне естественно, но доста-
точно нетривиально. Описание этого достойно отдельного раздела. В любом случае
следует действовать по правилу: если сомневаетесь, то расставляйте скобки в нуж-
ных местах.

В META можно упускать некоторые из операторов для сокращения записей, на-
пример, 2*x соответствует записи 2x. Но будьте осторожны, так как 1/2x это 0.5x,
что более естественно с точки зрения математики, но не программирования. В META

сначала обрабатываются числовые лексемы.

Набор стандартных вычислительных операций расширен с учётом специализа-
ции языка. В частности поддерживаются операции пифагорова сложения a++b =√

a2 + b2, пифагорова вычитания a+−+b =
√

a2 − b2), целочисленное деление div

и возведение в степень x∗∗y = xy.

В языке присутствуют операторы цикла, условных переходов и тому подобное.
Отличительной особенностью META является возможность решать линейные урав-
нения. Например, выражения вида C = 1/2[A,B] означает, что точка C находится
ровно посередине между точками AиB.

Программу metapost можно использовать в режиме калькулятора для вычисле-
ний на языке META. Это позволяет проверить правильность ваших предположений
относительно языка. Пример сеанса представлен ниже:

> mpost
This i s MetaPost , Vers ion 0 . 901 (Web2C 7 . 5 . 5)
∗∗\ r e l a x

∗a :=10 ;

∗b :=8 ;

∗c :=a+−+b ;

∗show c ;
>> 6
∗show (3− s q r t 5)/2 ;
>> 0 . 38197
∗show ang le (1 , s q r t 3) ;
>> 60 . 00008
∗show 2∗∗10 ;
>> 1024 . 00003
∗show i n f i n i t y ;
>> 4095 . 99998
∗show ep s i l o n ;
>> 0 . 00002
∗show i n f i n i t y− i n f i n i t y ;
>> 0
∗end
Transcr ipt wr i t t en on mpout . l o g .

После вывода приглашения ** следует набрать команду \relax. Далее можно вво-
дить команды MetaPost. Делайть это надо аккуратно, так как этот режим не под-
держивает «истории команд». В начале не предполагалось, что MetaPost можно
использовать и так тоже. С помощью команды show можно вывести результат на
экран. Закончить сеанс можно с помощью команды end.

Обратите внимание, что на просьбу вывести бесконечность (infinity) MetaPost
выдал 4095.99998— это максимальное значение, которое может принемать перемен-
ная типа numeric. Причём в процессе вычисления результат может привышать
«бесконечность», но ответ должен быть меньше или равен её, иначе будет выда-
на ошибка. Минимальный шаг изменения типа numeric равен epsilon, или точнее
1/256/256. При создании рисунка эти ограничения не существенны, так как диа-
пазон изменения чисел вполне велик, чтобы вместить все элементы. Но в любом
случае это тоже необходимо учитывать.

Если Вы хотите вычислить однострочное выражение, то на первоначальное при-
глашение ** можно ввести expr. В этом случае mpost считает файл expr.mf и на
любое ваше действие будет выдаваться ответ:

> mpost
This i s MetaPost , Vers ion 0 . 901 (Web2C 7 . 5 . 5)
∗∗ expr
(/ usr / l o c a l / t e x l i v e /2005/ texmf−d i s t /metafont/base / expr .mf
gimme an expr : 2( a+3b)−2b
>> 4b+2a
gimme an expr : 1/3 [ a , b ]
>> 0 . 33333b+0 . 66667a

1.5. Литература

Язык META, который используется в MetaPost за некоторыми исключениями пол-
ностью соответствует диалекту META, который используется в программе создания
шрифтов METAFONT.

Основной книгой по языку META является «Всё про METAFONT» Дональда Э.Кнута.
В 2003 году издательством Вильямс был выпущен русский перевод этой классиче-
ской книги (ISBN 5-8459-0442-0). Исходники англоязычного оригинала «The META-
FONT book» можно найти на любом CTAN архиве. Эта книга, как и другие произведе-
ния Д.Э.Кнута имеет несколько уровней сложностей. Даже с нулевым начальным
уровнем знания предмета Вы можете прочитать книгу полностью— её строение
это позволяет, но для дальнейшего продвижения эту книгу придётся перечитывать
не один раз. Каждое прочтение приносит новое понимание. Если Вы имеете хоть
какое-то отношение к программированию, то книги Д.Э.Кнута надо читать. Именно
благодаря подобным людям информатика может претендовать на фундаменталь-
ность.

На русском языке информацию о MetaPost можно найти в «Путеводителе по па-
кету LATEX и его графическим расширениям» М.Гуссенса, С.Ратца и Ф.Миттельбаха
от издательства Мир (ISBN 5-03-003388-2).

Все остальные источники в основном англоязычные. Прежде всего это «A User’s
Manual for MetaPost» Джона Хобби (Jhon D. Hobby) —файл mpman.pdf. Этот текст
можно найти в документации к дистрибутиву MetaPost. Этому тексту предшесто-
вало несколько «основополагающих» статей, которые при желании можно легко
найти в интернете. Очень качественным руководством пользователя отметился Ан-
дрэ Хек (André Heck) http://remote.science.uva.nl/~heck/Courses/mptut.pdf.
Так же представляет интерес книга «metafun» от Ганса Хагена (Hans Hagen) — она
находится в открытом доступе, имя файла metafun-p.pdf.

В интернете можно найти интересный ресурс под названием «Metapost : exemples»
от Vincent Zoonekynd http://zoonek.free.fr/LaTeX/. Это страничка с огромным
количеством простых примеров. Небольшое неудобство состоит в том, что это фран-

коговорящий ресурс, с другой стороны текста там немного. Зеркало расположено,
например, здесь http://tex.loria.fr/prod-graph/zoonekynd/metapost/metapost.html.

Ко всей перечисленной электронной документации можно получить доступ через
страничку CTAN, посвящённую MetaPost: http://www.tug.org/metapost.html

Глава 2
Базовые элементы

Определённо всё из чего-то состоит.
Элементарные кирпичики — вот что интересно.

Объяснять компьютеру что Вы хотите сделать гораздо сложнее, чем нарисовать
самому. Компьютеру надо объяснять абсолютно всё — телепатические способности
у машин на текущий момент отсутствуют. Но, объяснив один раз, все похожие
действия выполняются путём небольшой модификации уже готовых инструкций. В
результате потраченное на объяснение время себя полностью оправдывает, правда,
при этом требуются дополнительные «мозговые усилия».

2.1. Рисуем по точкам

Первое, что надо сделать перед созданием нового рисунка, это сделать его набро-
сок на миллиметровки. Допустим, необходимо нарисовать черепашку:

−10 −5 0 5 10

−5

0

5

Рис. 2.1. Черепашка

Делаем это, как можем, а затем очень подробно объясняем компьютеру, что мы
хотим от него. Инструкции очень простые:

% Файл coord.mp

% Черепашка

beginfig ( 1 ) ;
numeric u ; u:=5mm;
draw (−5u , 0 u){ dir 9 0 } . . { dir 0} (0 , 3 . 5 u ){ dir 0 } . .

{dir −90}(5u , 0 ) { dir −90}..{ dir 180}(0u ,−3.5u){ dir 1 8 0 } . .
{dir 90} cycle withpen pencircle scaled 0 . 4u ;

draw ( 4 . 5 u , 1 . 5 u ) . . ( 7 u , 1 . 5 u ) . . ( 8 u , 0 u ) . . ( 7 u ,−1.5u ) . .
( 4 . 5 u ,−1.5u) withpen pencircle scaled 0 . 4u ;

draw ( 6 . 5 u , . 5 u ) withpen pencircle scaled 0 . 8u ;
draw ( 6 . 5 u ,− .5u ) withpen pencircle scaled 0 . 8u ;
draw (−3.5u ,−2.5u ) . . ( −4 . 2u ,−4.5u ) . . ( −3 . 7u,−5u ) . .

(−2.4u,−4u ) . . (−2u ,−3.5u) withpen pencircle scaled 0 . 4u ;
draw (−3.5u , 2 . 5 u ) . . ( −4 . 2u , 4 . 5 u ) . . ( −3 . 7u , 5 u ) . . ( −2 . 4u , 4 u ) . .

(−2u , 3 . 5 u ) withpen pencircle scaled 0 . 4u ;
draw ( 3 . 5 u ,−2.5u ) . . ( 4 . 2 u ,−4.5u ) . . ( 3 . 7 u,−5u ) . . ( 2 . 4 u,−4u ) . .

(2u ,−3.5u) withpen pencircle scaled 0 . 4u ;
draw ( 3 . 5 u , 2 . 5 u ) . . ( 4 . 2 u , 4 . 5 u ) . . ( 3 . 7 u , 5 u ) . . ( 2 . 4 u , 4 u ) . .

(2u , 3 . 5 u ) withpen pencircle scaled 0 . 4u ;
draw (−5u , 0 . 5 u)−−(−6.5u,0)−−(−5u ,−0.5u)

withpen pencircle scaled 0 . 4u ;
draw (0u , 2 . 5 u ) . . ( 0 , 0 u ) . . ( 0 u ,−2.5u)

withpen pencircle scaled 0 . 3u ;
draw (−2.4u , 2 . 4 u ) . . (−2u , 0 u ) . . ( −2 . 4u ,−2.4u)

withpen pencircle scaled 0 . 3u ;
draw ( 2 . 4 u , 2 . 4 u ) . . ( 2 u , 0 u ) . . ( 2 . 4 u ,−2.4u)

withpen pencircle scaled 0 . 3u ;
draw (−3u , 1 . 5 u ) . . ( 0 , 1 u ) . . ( 3 u , 1 . 5 u )

withpen pencircle scaled 0 . 3u ;
draw (−3u ,−1.5u ) . . (0 , −1u ) . . ( 3 u ,−1.5u)

withpen pencircle scaled 0 . 3u ;
endfig ;

Рисуем по точкам, используя команду draw. Каждая точка задаётся парой чисел
«(x,y)» — x и y координаты соответственно. Точки соединяются либо прямыми ли-
ниями «- -», либо кривыми «. .». Кривые, соединяющие точки описываются полино-
мом Бернштейна третьей степени (Сергей Николаевич Бернштейн 1912). Часто их
называют кубическими кривыми Безье (Piere Bézier 1960).

Компьютер обязательно должен знать каким пером withpen pencircle и какой
толщины scaled 0.3u рисуется текущая линия. Такие объяснения могут показать-
ся избыточными, но всегда можно скопировать предыдущую инструкцию и попра-
вить её под свои нужды, а многословность позволяет легче читать код.

Далее этот рисунок можно использовать многократно, например, для составле-
ния какого-либо узора:

Рис. 2.2. Черепаший узор

Обратите внимание, что почти все числовые значения представляют из себя коэф-
фициент умноженный на параметр, например, «2.4u». Параметру «u» можно при-
своить числовое значение в миллиметрах (параметр «mm»). Есть несколько опреде-
лённых по умолчанию значений длины, например, «cm» соответствует сантиметру.

При программировании на META по мере возможности используйте параметри-
ческие зависимости. Нет необходимости «прибивать что-то гвоздями». В данном
случае наличие параметра позволяет легко изменить масштаб рисунка:

Рис. 2.3. Черепашья семья

Очевидно, что в предложенном решении одного параметра мало, так как «u»
контролирует не только геометрические размеры, но и размеры пера.

Точки, тоже можно представить как переменные, имеющие тип pair:

numeric u ;
u :=0.5mm;
pair A,B;
A:=(1u , 2 u ) ;B=(5u ,10u ) ;
draw A−−B withpen pencircle scaled 0 . 3u ;

Следует обратить внимание, что знак присвоения «:=» в случае переменной A и
знак равенства в случае переменной B здесь действует одинаково. Отличия воз-
никают, когда используется уникальная способность META воспринимать и решать
систему линейных уравнений. В дальнейшем в объяснениях буквы A и B будут
использоваться как переменные типа точка.

META позволяет создавать свои типы перьев, но для рисования простых рисунков
использовать что-то отличное от круглого пера (pencircle) нет особой необходи-
мости. Если вам хочется расширить ваши познания в этой области, то обратитесь
к любому руководству по METAFONT или MetaPost. Нет необходимости каждый раз
указывать каким пером следует рисовать. Достаточно выбрать какое-либо перо по
умолчанию с помощью команды pickup, например, так:

pickup pencircle scaled 0 . 2u ;

2.2. Пути

Инструкции draw позволяет рисовать сплошную линию. На её основе в MetaPost
создана команда для рисования стрелки drawarrow. Воспользуемся ей для изобра-
жения векторной суммы: Кроме, непосредственно, команды draw в этом рисунке

~F1

~F2

~F3

~F4

~F1
+

~F2
+

~F3
+

~F4

Рис. 2.4. Сложение сил.

необходимо использовать команду для вставки текстовых меток label, которая бу-
дет подробно разобрана позже:

% Файл path.mp

% Закон Ньютона - векторная сумма сил

beginfig ( 1 ) ;
numeric u ;
u := 1mm;
drawarrow (0 ,0)−−(20u ,30u) withpen pencircle scaled 0 . 3u ;
label . ul f t (btex \(\ vec {F}_1\) etex , 1 / 2 [ ( 0 , 0 ) , ( 2 0 u ,30u ) ] ) ;
drawarrow (20u ,30u)−−(40u ,30u) withpen pencircle scaled 0 . 3u ;
label . top (btex \(\ vec {F}_2\) etex , 1 / 2 [ ( 2 0u ,30u ) , ( 40 u ,30u ) ] ) ;
drawarrow (40u ,30u)−−(50u ,10u) withpen pencircle scaled 0 . 3u ;
label . urt (btex \(\ vec {F}_3\) etex , 1 / 2 [ ( 4 0u ,30u ) , ( 50 u ,10u ) ] ) ;

drawarrow (50u ,10u)−−(35u ,15u) withpen pencircle scaled 0 . 3u ;
label . l l f t (btex \(\ vec {F}_4\) etex , 1 / 2 [ ( 5 0u ,10u ) , ( 35u ,15u ) ] ) ;
drawarrow (0u , 0 u)−−(35u ,15u) withpen pencircle scaled 0 . 8u ;
drawarrow (0u , 0 u)−−(35u ,15u) withpen pencircle scaled 0 . 3u

withcolor white ;
draw (0u , 0 u) withpen pencircle scaled 2u ;
label . bot (btex \(\ vec {F}_1+\vec {F}_2+\vec {F}_3+\vec {F}_4\) etex ,

1 /2 [ ( 35u ,15u ) , ( 0 u , 0 u ) ] ) rotatedaround
( 1/2 [ ( 35 u ,15u ) , ( 0 u , 0 u ) ] , angle ( 3 5 , 1 5 ) ) ;

endfig ;

Конструкция вида «1/2[A,B]» имеет тип pair и равна точке, расположенной ровно
по середине между точками A и B. Точно так же можно выбрать точку на линии
AB, но делящую эту линию 1 к 2: «1/3[A,B]», или 1 к 4: «0.2[A,B]»

Кроме drawarrow в MetaPost определена команда drawdblarrow, которая рисует
кончик стрелки на обоих концах пути.

Команды, типа draw, работают с объектами path (путь). Путь — это набор из
точек (тип pair) с описанием как эти точки друг с другом соединяются. Мини-
мальный путь, это одна точка. При рисовании такого пути в рисунке остаётся от-
печаток в форме пера. Во всех приведённых здесь примерах перо имеет круглую
форму pencircle, поэтому в этом случае возникает просто точка. Часто бывает
удобно создать переменную типа path:

% Файл path.mp

% Рис к 1.6.8 б) нить пропущенная через изогнутую трубу

beginfig ( 2 ) ;
numeric u ;
u = 0 .8mm;
path p ;
p :=(5u , 0 u)−−(20u , 0 u){ dir 0 } . . ( 2 0 u ,20u ) . .

{dir 0}(20u , 0 u)−−(35u , 0 u ) ;
cutdraw p withpen pencircle scaled 1 . 5u ;
draw p withpen pencircle scaled 1u withcolor white ;
draw p withpen pencircle scaled 0 . 3u

dashed evenly scaled 1/2u ;
drawarrow (−10u , 0 u)−−(0u , 0 u)

withpen pencircle scaled 0 . 3u ;
draw (−10u , 0 u) withpen pencircle scaled 1u ;
draw (−10u , 0 u)−−(5u , 0 u) withpen pencircle scaled 0 . 3u ;
drawarrow (30u ,10u)−−(50u ,10u)

withpen pencircle scaled 0 . 3u ;
draw (30u ,10u) withpen pencircle scaled 1u ;
draw (35u , 0 u)−−(45u , 0 u) withpen pencircle scaled 0 . 3u ;
label . top (btex \(\ vec {v}\) etex ,(−5u , 0 u ) ) ;
label . top (btex \(2\ vec {v}\) etex , ( 4 0 u ,10u ) ) ;

endfig ;

В данном примере требовалось изобразить трубу изогнутую буквой «О», сквозь ко-
торую продета нить. Для этого определяем путь «p». Затем этот путь используется
в трёх командах как переменная: рисуется толстая чёрная труба шириной «1.5u»,
внутри неё рисуется более тонкая шириной «1u» белого цвета (withcolor white) —
получается полая труба, а затем внутри трубы отрисовывается нить, причём нить
рисуется пунктиром (dashed evenly).

~v

2~v

Рис. 2.5. Изогнутая труба.

Первый и последний участок изогнутой трубы прямые линии, поэтому соедине-
ние с первой и последней точкой описывается как «- -». Чтобы нарисовать кривую,
похожую на круг, достаточно двух точек:

numeric u ,R; u=1mm;R=10u ;
pair A,B; A:=(−R, 0 ) ;B:=(0 ,R) ;
path P; P:=A . . B . . cycle ;
draw P withpen pencircle 1u ;

В этом примере определяются две точки A и B. Путь P строится по этим точкам,
при этом выходя из точки A, мы попадаем в точку B, а затем снова в точку A:
команда cycle позволяет создавать замкнутую кривую. Настройки META по умол-
чанию таковы, что получившаяся кривая достаточно хорошо совпадает с точной
окружностью. Чтобы лучше совпадать с окружностью нужно больше «опорных»
точек. Просто, обычно, для рисунков точности построения по двум точкам хватает,
но всё-таки надо осознавать, что отличие есть.

Для уточнения пути в некоторых точках можно указать под каким углом должна
подходить кривая к этой точке. Инструкция вида «{dir α}» если она находится пе-
ред точкой, указывает под каким углом кривая должна подходить, а после точки —
под каким углом кривая должна уходить. α — угол в градусах от оси абсцисс. Для
указания направления можно так же воспользоваться сокращениями left, right,
up и down, которые означают, соответственно, влево, вправо, вверх и вниз.

Существует несколько типов соединений между точками. Два из них «..» и «- -»
уже изучили. Из других типов полезен тип «&»— «сращивание» (объединение двух
путей без влияния друг на друга) и «- - -» — «натянутая» линия (то же, что и «- -»,
но влияет на соседние соединительные участки).

Обратите внимания на команду cutdraw. Так как MetaPost рисует с помощью
перьев, то в случае круглого пера (pencircle), концы линий также округлые. В

случае когда необходимо «обрубить» концы, избавиться от округлостей на конца
линий, используется эта инструкция. Следует отметить, что эта команда «подчист-
ки» концов использует другую более общую команду cutoff ("точка","угол");

2.3. Вставка текста

METAFONT был создан как специализированный инструмент для создания шриф-
тов. MetaPost создавался как инструмент для рисования любых изображений, по-
этому в него был встроен довольно мощный механизм вставки текста. Работая сов-
местно с LATEX, MetaPost позволяет использовать всю мощь текстового процессора
для создания надписей. Кроме всего прочего, иметь в рисунках те же шрифты, что
и в тексте просто красиво.

Разберём это на примере треугольник Паскаля1:

10:
1 11:

1 2 12:
1 3 3 13:

1 4 6 4 14:
1 5 10 10 5 15:

1 6 15 20 15 6 16:
1 7 21 35 35 21 7 17:

1 8 28 56 70 56 28 8 18:
1 9 36 84 126 126 84 36 9 19:

1 10 45 120 210 252 210 120 45 10 110:

1 11 55 165 330 462 462 330 165 55 11 111:

1 12 66 220 495 792 924 792 495 220 66 12 112:

1 13 78 286 715 1287 1716 1716 1287 715 286 78 13 113:

1 14 91 364 1001 2002 3003 3432 3003 2002 1001 364 91 14 114:

(a + b)n =
=

∑

C
k
na

k
b
n−k

Т
ре
уг
ол
ьн
ик

П
ас
ка
ля

Рис. 2.6. Треуголник Паскаля.

Так как нижние строки из-за увеличения разрядности чисел расползались, то для
представления появилась необходимость уменьшить их геометрические размеры.
В коде так же использовались циклы, которые будут разобраны позже. Команда
show позволяет отлаживать код, выдавая значения переменных на экран во время
компиляции рисунка с помощью mpost.

% Файл pic.mp

% Треугольник Паскаля

beginfig ( 5 ) ;
numeric u ;
u = 1 .mm;
numeric dy , dx , x , y , n [ ] [ ] , i , j , sy , ds , n l a s t ;
dy:=5u ; dx:=5u ; x=0;y=0;

1Треугольник Паскаля — треугольная числовая таблица для составления биномиальных ко-

эффициентов. По боковым сторонам треугольника стоят единицы, внутри треугольника числа

образуются сложением двух чисел, стоящих над данным.

ds =0.04 ; sy =0.032 ; n l a s t =14;
picture z ;
for i :=0 upto n l a s t :

dy:=dy∗(1−sy ) ;
y:=y−dy ;
for j :=0 upto i :

i f ( j =0) or ( j=i ) :
n [ i ] [ j ] :=1 ;

else :
n [ i ] [ j ] :=n [ i −1] [ j−1]+n [ i −1] [ j ] ;

f i
% show i,j,n[i][j];

z := thelabel (decimal (n [ i ] [ j ] ) , ( 0 , 0 ) ) ;
x:=dx∗( j−i / 2 ) ;
label ( z scaled (1−ds∗ i ) , ( x , y ) ) ;

endfor
z := thelabel . l f t (decimal ( i )&" : " , ( 0 , 0 ) ) ;
label ( z scaled (1−ds∗ i ) , ( dx∗(− n l a s t /2−1) ,y ) ) ;

endfor
label . rt (btex \ ( ( a+b)^n=\) etex , ( 5 u,−5u ) ) ;
label . rt (btex \(=\sum C^k_na^kb^{n−k}\) etex ,

(10u,−10u ) ) ;
label . l f t (btex Треугольник Паскаля etex

rotated 56 ,(−5u,−20u ) ) ;
endfig ;

Команда decimal делает из переменной типа numeric строку. Две строки можно
слить с помощью операнда «&».

Для того чтобы вставить текстовую метку в рисунок, необходимо получить на
входе текстовую строку, которую, возможно, надо будет преобразовать с помощью
LATEX, и точку где эту строку следует расположить:

label ( " t ext ␣ s t r i n g " ,A) ;

Полезно ещё уточнить с какой стороны от указанной точки расположить текстовую
метку. Уточнение производится с помощью суффикса, который добавляется мак-
росу label через точку. Всего существует восемь стандартных суффиксов: «.rt» —
расположить справа, «.lft» — слева, «.top» — сверху, «.bot» — снизу, «.llft» — распо-
ложить снизу и слева по диагонали, «.lrt» — снизу и справа, «.ulft» — сверху и слева,
«.urt» — сверху и справа.

Если label передаётся просто строка (тип string), то текст обрабатывается си-
лами MetaPost и всё, что выходит за пределы ASCII-таблицы с большой вероятно-
стью не отобразится. Для того чтобы строка была обработана LATEX, необходимо это
указать с помощью разделителей btex и etex. Строка между этими разделителями
обрабатывается LATEX, при этом в качестве заголовка используются инструкции пе-

речисленные в начала mp-файла между verbatimtex и etex. Таким образом можно
использовать кириллицу и любую конструкцию, которую понимает LATEX.

В некоторых случаях команде label удобно передавать не строку, а картинку
picture. В нашем случае это было необходимо, так как надпись надо было мас-
штабировать. Объект picture представляет из себя совокупность примитивов типа
путей и точек, поэтому его можно трансформировать. Команда thelabel создаёт
такую картинку. Объект, заключённый между btex и etex так же является кар-
тинкой.

2.4. Заливка

Кроме рисования кривых часто бывает необходимо закрасить какую-либо за-
мкнутую область.

α

~v

~v

~v

Рис. 2.7. «Струя воды».

Для этого существует команда fill. На вход команды fill подаётся объект типа
path, при этом путь должен быть замкнутым, то есть оканчиваться командой cycle.
Команда cycle автоматически замыкает кривую.

% Файл pic.mp

% Рис к 1.6.20 а) струя под углом , разбивающаяся о пол

beginfig (10) ;
numeric u ;
u = 1mm;
numeric s t ; s t :=u/( sqrt 3 ) ;
path p ;
p :=(0 ,10u)−−(18u ,10u){ dir 0 } . . { dir 30}(22u ,10u+2s t)−−

(50u ,10u+30s t ){ dir −90}..{ dir −90}(60u , 5 u+20s t)−−
(42u , 5 u+2s t ){ dir −150}..{ dir 0}(42u , 5 u)−−
(60u , 5 u){ dir −120}..{ dir −120}(60u , 0 u)−−
(0u , 0 u){ dir 6 0 } . . { dir 60} cycle ;

f i l l p withcolor 0 . 8 white ;
draw ( (40u , 5 u)+10u∗dir 30){dir −60}..{ dir −90}(50u , 5 u ) ;
label (btex \(\ alpha \) etex , ( 5 3 u , 8 u ) ) ;
draw (0u,0)−−(60u , 0 u) withpen pencircle scaled 0 . 8u ;

numeric R, alpha ;R:=12u ; alpha=30;
drawarrow (25u+25u , 5 u+25s t)−−

( (25u+25u , 5 u+25s t )−R∗dir alpha )
withpen pencircle scaled 0 . 6u ;

label . ul f t (btex \(\ vec {v}\) etex , 1 / 2 [ ( 2 5u+25u , 5 u+25s t ) ,
( (25u+25u , 5 u+25s t )−R∗dir alpha ) ] ) ;

drawarrow (20u , 5 u)−−(20u−R,5 u)
withpen pencircle scaled 0 . 6u ;

label . top (btex \(\ vec {v}\) etex , ( 2 0 u−R/2 ,5u ) ) ;
drawarrow (45u , 2 . 5 u)−−(45u+R, 2 . 5 u)

withpen pencircle scaled 0 . 6u ;
label . l f t (btex \(\ vec {v}\) etex , ( 4 5 u , 2 . 5 u ) ) ;

endfig ;

Для fill существует противоположная по смыслу команда unfill, которая, со-
ответственно, убирает заливку в выбранной окрестности. Если вы планируете ис-
пользовать эти команды, то изучите соответствующий раздел «Всё про METAFONT»
Кнута, так как влияние этих команд друг на друга не совсем тривиально. Дело в
том, что при наложении друг на друга двух заливок в месте пересечения образуется
«двойной слой краски» и для того чтобы убрать его необходимо дважды вызывать
команду unfill. Команда unfill действует схоже, только число «слоёв краски»
становится отрицательным. Для сведения всё к ситуации когда краска либо есть
(один слой), либо её нет используется команда «выравнивания» cullit.

Обратите внимание, что точку можно задать не только парой чисел: «(x,y)» —
Декартова система координат, но и радиусом с направлением «R∗dir α»— полярные
координаты.

2.5. Цвета

Ещё одно важное отличие MetaPost от METAFONT это наличие цвета. Работа с
цветом, обычно сложнее чем работа с чёрно-белым рисунком. До сих пор проце-
дура переноса электронного цветного рисунка на бумагу не является тривиальной.
Но время идёт, и цветные принтеры становятся всё доступней. Кроме того, хоро-
шие чисто электронные тексты, не привязанные к твёрдой копии, так же становятся
весьма распространёнными. Поэтому если необходимо, то следует пользоваться цве-
том, естественно, если вы знаете что делать. В случае простого рисунка цвет, как
правило, только отвлекает.

В MetaPost управление цветом реализовано на довольно низком уровне. Цвет
определяется объектом типа color и представляет из себя тройку чисел принима-
ющих значение от 0 до 1: «(r,g,b)», где r соответствует красной, g — зелёной, а b —
голубой компоненте. Существует пять предопределённых цветовых константы: red
(1,0,0), green (0,1,0), blue (0,0,1), а так же black (1,1,1) и white (0,0,0).

Нарисовать объект, выбранным цветом, можно с помощью инструкции withcolor

за которой следует сам цвет. Цвета можно складывать, вычитать, умножать на
число. Пользуясь базовыми определениями можно создать более сложные объекта.
Например как этот спектр:

Рис. 2.8. Спектр.

Код, создающий эту картинку, далёк от совершенства, но он простой и его можно
улучшать в случае необходимости:

% Файл colors.mp

def s p e c t r l i n e (expr i c , c , l ,w, dw) =
begingroup

save i , i f i r s t , i l a s t , i s t ep , i c c ;
color i c c ;
i f ( i c >0): i s t e p :=1; i f i r s t :=0; i l a s t =255;
else : i s t e p :=−1; i f i r s t =0; i l a s t :=−255; f i ;
i f (abs ( i c )=1) : i c c :=red ; f i ;
i f (abs ( i c )=2) : i c c :=green ; f i ;
i f (abs ( i c )=3) : i c c :=blue ; f i ;
for i := i f i r s t s tep i s t e p until i l a s t :

draw ( ( 0u , 0 u)−−(0u , l ) ) shifted (w+(dw/2∗abs ( i ) , 0 ) )
withpen pencircle scaled dw withcolor ( c+i /255∗ i c c ) ;

endfor ;
endgroup ;

enddef ;

% спектр

beginfig ( 1 ) ;
color cur r en t ;
u:=1mm;w:=200u ;dw:=w/512 ; l :=5u ;
cu r r en t := (1 , 0 , 0 ) ;
s p e c t r l i n e (2 , current , l , ( 1 / 4w, 0 u ) ,dw)
cur r en t := (1 , 1 , 0 ) ;
s p e c t r l i n e (−1 , current , l , ( 1 / 2w, 0 u ) ,dw)
cur r en t := (0 , 1 , 0 ) ;
s p e c t r l i n e (3 , current , l , ( 3 / 4w, 0 u ) ,dw)
cur r en t := (0 , 1 , 1 ) ;
s p e c t r l i n e (−2 , current , l , (w, 0 u ) ,dw)

endfig ;

В примере используется макрос spectrline для уменьшения размера кода.

23



Глава 3
Начала автоматизации

Компьютер не умеет читать ваши мысли,
зато неукоснительно следует инструкциям.

До сего момента мы концентрировались на том, как объяснить компьютеру, что-
бы он сделал то или иное движение. Теперь воспользуемся способностью компьюте-
ра помнить предыдущие действия и извлекать их из памяти по мере необходимости.
Автоматизация рутинных процедур это то, для чего компьютер и предназначены.
Практиковаться в автоматизации следует постоянно. Не смотря на затраченное на
обучение время, в результате время же и экономится.

3.1. Объекты picture

В процесс повествования объект picture или картинка уже упоминался. Кар-
тинка представляет из себя совокупность путей и точек которую можно подвергать
трансформации. В уже существующие картинки можно добавлять пути, замкнутые
области и другие картинки.

Для начала опять же воспользуемся миллиметровкой для отрисовке какого-либо
рисунка, например, ракеты:

fi
r
e
r
o
c
k
e
t

rocket

fi
r
e

a
n
d

r
o
c
k
e
t

Рис. 3.1. Ракета

Ракета может быть без выхлопа (rocket) и c выхлопом (firerocket). В процессе
создания firerocket был использован рисунок самого выхлопа (fire).

% Файл picture .1.mp

% Ракета без выхлопа 10x12 Центр у стабилизаторов

picture r o cke t ;
r o cke t :=nullpicture ;
addto r o cke t contour (−2,−1)−−(−2,6)−−(0,10)−−(2,6)−−

(2,−1)−−cycle withpen pencircle scaled 0 . 4 withcolor white ;
addto r o cke t doublepath (−2,−1)−−(−2,6)−−(0,10)−−(2,6)−−

(2,−1)−−cycle withpen pencircle scaled 0 . 5 ; % Корпус

addto r o cke t contour (−2,2.5)−−(−4,1)−−(−4.5,−3)−−
(−2,−3)−−cycle withpen pencircle scaled 0 . 4 withcolor white ;

addto r o cke t doublepath (−2,2.5)−−(−4,1)−−(−4.5,−3)−−
(−2,−3)−−cycle withpen pencircle scaled 0 . 5 ; % левая дюза

addto r o cke t contour (2 ,2.5)−−(4 ,1)−−(4.5 ,−3)−−(2 ,−3)−− cycle
withpen pencircle scaled 0 . 4 withcolor white ;

addto r o cke t doublepath (2 ,2.5)−−(4 ,1)−−(4.5 ,−3)−−
(2,−3)−−cycle withpen pencircle scaled 0 . 5 ; % правая дюза

addto r o cke t doublepath (0 ,2.5)−−(0 ,−3)
withpen pencircle scaled 0 . 8 ; % центральная дюза

% выхлоп

picture f i r e ;
f i r e :=nullpicture ;
addto f i r e doublepath (0,−4)−−(0,−6)

withpen pencircle scaled 0 . 3 ; %выхлоп 1

addto f i r e doublepath (−1.5 ,−4)−−(−1.5 ,−6)
withpen pencircle scaled 0 . 3 ; %выхлоп 2

addto f i r e doublepath (1.5 ,−4)−−(1.5 ,−6)
withpen pencircle scaled 0 . 3 ; %выхлоп 3

addto f i r e contour (−2.5 ,−6.5){ dir 135} . . (−4 ,−8) . .
{dir 50}(−1.2 ,−8.2){ dir −110}. . (0 ,−10) . .
{dir 110}(1 .2 ,−8.2){ dir −50} . . (4 ,−8) . .
{dir −135}(2.5 ,−6.5)−−cycle withpen pencircle scaled 0 . 4
withcolor white ;

addto f i r e doublepath (−2.5 ,−6.5){ dir 135} . . (−4 , −8) . .
{dir 50}(−1.2 ,−8.2){ dir −110}. . (0 ,−10) . .
{dir 110}(1 .2 ,−8.2){ dir −50} . . (4 ,−8) . . {dir −135}(2.5 ,−6.5)
withpen pencircle scaled 0 . 3 ; % облако

% ракета и выхлоп

picture f i r e r o c k e t ;
f i r e r o c k e t := rocke t ;

addto f i r e r o c k e t also f i r e ;

Прежде чем к картинке что-то добавить её необходимо инициализировать. В MetaPost
есть две определённые по умолчанию картинки: nullpicture—пустая картинка и
currentpicture— текущая картинка. Пользуясь последней переменной, можно в
любой момент сохранить результаты промежуточной отрисовки. Добавление эле-
ментов к картинке производится с помощью инструкции addto далее указывается
картинка к которой добавляется тот или иной элемент. Путь добавляется с помо-
щью инструкции doublepath, замкнутая область с помощью инструкции contour,
а другая картинка с помощью инструкции also.

Ранее был создан рисунок черепашки. Для его обозначения была выбрана пере-
менная Turtle. Теперь с ней можно поработать, как с единым элементом, например,
для иллюстрации задачи: «Черепашки расположены в углах правильного треуголь-
ника со стороной a и всегда ползут в направлении своей соседки против часовой
стрелки со скоростью v. Когда они встретятся?»

O

Рис. 3.2. Встреча в центре

Картинку можно отобразить с помощью команды draw. Над картинкой можно
производить различные преобразования. В данном случае картинка поворачива-
лась, масштабировалась и сдвигалась.

% Файл pic.mp

beginfig (17) ;
numeric u ; u = 0 .8mm;
numeric dphi ; dphi=20;
draw 30u∗dir (90+dphi)−−30u∗dir (210+dphi)−−

30u∗dir (330+dphi)−−cycle dashed evenly scaled 1u ;
draw Turt le rotated (−120+dphi ) scaled 1u

shifted (30u∗dir (90+dphi ) ) ;
draw Turt le rotated dphi scaled 1u

shifted (30u∗dir (210+dphi ) ) ;
draw Turt le rotated (120+dphi ) scaled 1u

shifted (30u∗dir (330+dphi ) ) ;

endfig ;

Обратите внимание, что линия, соединяющая черепах нарисована пунктиром. Опре-
делённая по умолчанию переменная evenly тоже является картинкой, поэтому её
можно масштабировать с помощью декларации scaled. То есть, если вам нужен
более широкий шаг пунктира, то вместо масштаба 1u можно указать 2u. Если вас
не устраивает где располагаются штрихи у штриховки, то можно воспользоваться
декларацией сдвига shifted.

Кроме шаблона evenly в MetaPost определён шаблон withdots, который позво-
ляет рисовать кривую с помощью точек.

Вы можете определить свой шаблон для пунктира примерно следующим образом:

picture dash_center ;
dash_center :=dashpattern (on 3 of f 1 . 5 on 0 . 5 of f 1 . 5 ) ;
draw 30u∗dir (90+dphi)−−30u∗dir (210+dphi)−−

30u∗dir (330+dphi)−−cycle dashed dash_center scaled 1u ;

Функция dashpattern принимает список on/off с числовой информацией в ка-
кой момент рисовать/не рисовать. В этом примере определён шаблон для штрих-
пунктирной линии, которая обычно используется для обозначения симметрии.

3.2. Трансформация

К задаче №3 варианта ГГФ-51в требовалось изобразить L-образную трубку с во-
дой. По условию трубка сначала стояла вертикально, а потом была положена на
стол.

h

H

P = ?

~g

d

h

H

~g

Рис. 3.3. Изогнутая пробирка на столе

Вовсе необязательно работать в трёхмерном редакторе чтобы схематично это
изобразить. Ниже идёт код, который рисует вертикально стоящую пробирку с раз-
мерами, а затем наклоняет её.

% Файл transform.mp

% пример использования slanted

beginfig ( 1 ) ;
numeric u ;

u = 0 .8mm;
% пробирка

cutdraw (0u , 0 u)−−(20u , 0 u)−−(20u ,30u){ dir 9 0 } . .
{dir −90}(17u ,30u)−−(17u , 3 u)−−(0u , 3 u)
withpen pencircle scaled 0 . 5u ;

drawdblarrow (23u ,10u)−−(23u , 1 u ) ;
label . rt (btex \(h\) etex , 1 / 2 [ ( 2 3u ,10u ) , ( 23u , 1 u ) ] ) ;
drawdblarrow (30u ,30u)−−(30u , 1 u ) ;
label . l f t (btex \(H\) etex , 1 / 2 [ ( 3 0u ,30u ) , ( 30 u , 1 u ) ] ) ;
picture Base ;
Base :=currentpicture ; % запоминаем

c l e a r i t ; % очищаем текущую картинку

% рисуем воду когда пробирка будет наклонена

f i l l (15u , 0 u)−−(20u , 0 u)−−(20u ,20u)−−(17u ,20u)−−
(17u , 3 u)−−(15u , 3 u)−−cycle withcolor 0 . 7white ;

draw Base ;
draw (12u ,20u)−−(20u ,20u ) ;draw (12u ,10u)−−(20u ,10u ) ;
drawdblarrow (14u ,20u)−−(14u ,10u ) ;
label . l f t (btex \(d\) etex , ( 1 4 u ,16u ) ) ;
picture Slant ;
S lant=currentpicture ; % запоминаем

c l e a r i t ; % очищаем текущую картинку

% рисуем воду когда пробирка стоит

f i l l (5u , 0 u)−−(20u , 0 u)−−(20u ,10u)−−(17u ,10u)−−
(17u , 3 u)−−(5u , 3 u)−−cycle withcolor 0 . 7white ;

% отрисовываем пробирку

draw Base ;
% отрисовываем пробирку и наклоняем её

draw Slant yscaled 2/3 slanted 1/2 shifted (40u , 0 u ) ;
endfig ;

В примере применяется возможность сохранить текущее состояние с помощью currentpicture,
а так же возможность полностью очистить текущую картинку с помощью инструк-
ции clearit.

h

H

h

Hd

Рис. 3.4. Изогнутая пробирка

Наклон вертикально стоящей пробирки происходит с помощью масштабирования
yscaled и, собственно, наклона slanted.

MetaPost поддерживает следующие базовые линейные преобразования:

(x, y) shifted (a, b) = (x + a, y + a)
(x, y) scaled s = (sx, sy)
(x, y) xscaled s = (sx, y)
(x, y) yscaled s = (x, sy)
(x, y) slanted s = (x + sy, y)
(x, y) rotated θ = (x cos θ − y sin θ, x sin θ + y cos θ)
(x, y) zscaled (a, b) = (xa − yb, xb + ya)

Кроме перечисленных базовых преобразований полезными для использования яв-
ляются макросы (x, y) rotatedaround ((a, b), θ)— поворот вокруг точки (a, b) на угол
θ и (x, y) reflectedabout (z1, z2)— отражение относительно линии, проходящей через
точки z1 и z2.

MetaPost поддерживает объекты типа transform, то есть можно определить лю-
бое необходимое для вас преобразование, чтобы использовать его в дальнейшем.

transform t ;
t := identity yscaled 2/3 slanted 1/2 shifted (40u , 0 u)
draw Slant transformed t ;

Используемая при описании преобразования t константа identity тоже является
преобразованием. identity— это «пустое» преобразование, то есть преобразование,
которое ничего не делает.

3.3. Циклы и условные операторы

Циклы и условные операторы в META отличаются от того, что обычно есть в
других языках программирования. Цикл не просто повторяет перечисленные в те-
ле цикла инструкции — он дублирует текст, то есть внутри цикла не обязательно
должна находиться синтаксически законченная конструкция. Это же относится и
к условным операторам.

«Шарик с постоянной скоростью движется в вдоль спицы, которая с вращается
с постоянной угловой скоростью. Требуется изобразить траекторию шарика.»

Для изображения траектории надо построить минимодель явления и задать фи-
зические параметры: поступательную скорость вдоль спицы v, угловую частота w и
начальные условия r и ϕ. Сама траектория создаётся с помощью следующего кода:

% Файл cycle.mp

v:=27u ;w:=360;N:=2 . 1 ; phi :=45; r :=5u ; n :=100;
path p ; pair O;
O:=( r ∗cosd ( phi ) , r ∗sind ( phi ) ) ;
p:=O for i=0 upto n :

~u

ω

Рис. 3.5. Спираль

. . ( ( r+v∗N∗ i /n)∗ dir(−w∗N∗ i /n)+phi ) )
endfor ;

draw p withpen pencircle scaled 0 . 5u
dashed evenly scaled 1u ;

Декларация upto это сокращение для step 1 until. Аналогично downto является
сокращением для step -1 untull

Формальный синтаксис цикла представлен ниже:

for i=x1 s tep x2 until x3 : text ( i ) endfor

Это одна из форм, которая поддерживается META. Ещё одна форма, представляет
бесконечный цикл:

forever : " текст " endfor

Для того чтобы выйти из подобного цикла необходимо воспользоваться конструк-
цией вида:

e x i t i f ( "булево␣выражение" )

Булево выражение, это может быть переменная типа boolean (true/false) или ре-
зультат сравнения чисел, точек, путей или преобразований. Операторы сравнения
совпадают с операторами сравнения языка «C» за исключением оператора равен-
ства «=» и оператора неравенства «<>». Выражение можно инвертировать с по-
мощью приставки not и объединить с другим с помощью приставок and или or.

Формальный синтаксис условного оператора представлен ниже:

i f ( "булево␣выражение1" ) : " текст1"
e l s e i f ( "булево␣выражение2" ) : " текст2"
else : " текст3" f i

Воспользуемся циклами для изображения циклоиды— траектория точки на ка-
тящемся колесе.

Рис. 3.6. Циклоида

Обратите внимание, что в конце цикла или условного оператора нет необходимо-
сти ставить «;» это позволяет использовать их довольно изощрённым образом.

% Файл cycle.mp

% Рис к задаче 1.5.8 (20x120) - циклоида

beginfig ( 1 ) ;
numeric u ; u = 0 .8mm;
numeric R; R=10u ;
path p , cy c l ;
p:=(−R,0u ) . . ( R, 0 u ) . . cycle ;
% колесо

draw p withpen pencircle scaled 0 . 3u
dashed withdots scaled 0 . 5u ;

numeric j , n , v ,w, phi , nsteps ;
j =0;n=100;v=109.8u ;w=−(v/R)∗180/3 . 14 ; phi =180; nsteps =4;
numeric r , i ;
for i :=0 upto 2∗ nsteps :

r :=R−1/nsteps ∗R∗ i ;
% метки

for j :=0 step n/4 until n :
draw ( j ∗(v/n)+r ∗cosd ( j ∗(w/n)+phi ) , r ∗sind ( j ∗(w/n)+phi ) )

withpen pencircle scaled 1u ;
endfor ;
% траектория меток

cyc l := for j :=0 upto n :
i f j <>0:.. f i

( j ∗(v/n)+r ∗cosd ( j ∗(w/n)+phi ) , r ∗sind ( j ∗(w/n)+phi ) )
endfor ;
draw cyc l dashed evenly scaled 1/2u

withcolor (max(1− i / nsteps , 0 ) ∗ red+
min( i / nsteps ,2− i / nsteps )∗green+
max( i / nsteps −1 ,0)∗blue ) ;

endfor ;
endfig ;

В этом коде выражение if j<>0:..fi использовалось для того, чтобы перед первой
точкой пути, описывающем циклоиду, не было декларации соединения. Я не знаю
какой из «популярных» на сегодня языков обладает такой способностью.

3.4. Макросы

Пользовательские функции в META фактически заменяются макросами. Как след-
ствие функции могут вернуть любую конструкцию от числа до картинки.

Один из моих ранних рисунков на META был «взрыв» в стакане. Требовалось изоб-
разить траекторию «осколков» которые летят по параболе и найти самую дальнюю
точку, которую достигают осколки при таком «взрыве».

r

h

R

~g

Рис. 3.7. Взрыв в стакане

Была написана процедура, которая рисовала параболу по переданным парамет-
рам. Вызов выглядел примерно следующим образом:

Parabola_dashed (0u , 0 u ,−1.25angle (20/ sqrt ( 8 ) ,
10∗ sqrt ( 8 ) ) , 10∗ sqrt (8 )∗u , 0 , 1 0 0 , 1 ) ;

Сам макрос для отрисовки параболы представлен ниже.

% Файл macros.mp

% Рисует параболу из точки (x,y) (полёт камня) штриховая

%линия. В качестве параметров передаётся (x,y), ang-угол,

%vel-скорость (100), %from,to - откуда и до куда рисовать

%параболу в процентах [0,100], mag - увеличение (0.8u)

%Для простоты g=10

def Parabola_dashed (expr x , y , ang , ve l , from , to ,mag) =
path p ;
numeric t , g , n ;
picture dash_one ;
dash_one:=dashpattern (on 2mag of f 2mag ) ;

n=100;% число шагов

g=10. ;
t :=(2∗ ve l ∗sind ( ang )∗ from )/ ( g∗n ) ;
p :=( ve l ∗cosd ( ang )∗ t ∗mag , ( ve l ∗sind ( ang )∗ t−g∗ t ∗ t /2)∗mag ) ;
for i=from+1 upto to :

t :=(2∗ ve l ∗sind ( ang )∗ i ) / ( g∗n ) ;
p:=p . . ( ve l ∗cosd ( ang )∗ t ∗mag ,

( ve l ∗sind ( ang )∗ t−g∗ t ∗ t /2)∗mag ) ;
endfor ;
draw p shifted (x∗mag , y∗mag) dashed dash_one ;

enddef ;

Не самое удачное решение, но оно выполняло то, что от него требовалось. В по-
добных случаях лучше чтобы в результате деятельности макроса оставался объект,
который потом можно нарисовать с помощью команды draw и трансформировать по
мере необходимости. Тогда функции передавалось бы гораздо меньше параметров,
что значительно всё упрощает. Например, вызов для рисования пружинки, которая
необходима в следующем примере, использует всего три входных параметра:

draw Spr ing (25u , 1 . 2 u , 2 5 ) rotated −90
shifted (−12.5u,−20u) withpen pencircle scaled 0 . 1u ;

«Два тела, соединённые пружинкой, висят в поле тяжести на нитях, образующих
угол в 90◦. В какой-то момент нить, крепящую конструкцию к потолку, разрывают.»
Надо нарисовать пружинку. Изображение пружинка может пригодится много где

90◦
~g

Рис. 3.8. Пружинка

ещё, поэтому оно было оформлено как макрос.

% Файл macros.mp

% Создаёт пружину, высоты h, радиуса r, с числом витков n

%(0,0) - в основании пружины

vardef Spr ing (expr h , r , n ) =
begingroup save i ;

(0 ,0)−−(0 ,− r /2+0.5h/n){ dir 180}
for i=h/n step h/n until h :

. . tension 1 . 2 . . ( − r , i−h/n ) . . tension 1 . 2 . .
(0 , r/2+i −0.5h/n ) . . tension 1 . 2 . . ( r , i ) . .
tension 1 . 2 . . (0 ,− r/2+ i +0.3h/n){ dir 180}

endfor−−(0,h )
endgroup

enddef ;

Обратите внимания на инструкцию tension—натяжение. Она говорит с какой «си-
лой» надо «натянуть» соединение между точками. Значение 1.2 означает, что это
следует сделать чуть потуже, чем обычно. С помощью этой инструкции описыва-
ется соединение между точками в определении пути типа «натянутая прямая»:

def −−− = . . t en s t i on in f in i ty . . enddef ;

Если отрисовка параболы является аналогом процедуры, то создание пружины
аналогом функции. Вызовы begingroup endgroup позволяют обособить вычисле-
ния, проводящиеся между ними, от «внешнего мира». С помощью команды save

можно защитить переменные внутри группы— «сохранённые» таким образом пере-
менные восстанавливают свои значения после выхода за пределы endgroup.

Параметры, которые передаются внутрь макроса перечисляются после деклара-
ции expr. Чтобы что-то вернуть в результате исполнения макроса, возвращаемое
выражение надо поместить в конце макроса без завершающего символа «;».

Отличие vardef от def заключается в том, что в случае def в качестве названия
макроса передаётся «символьная лексема», а в случае vardef «объявляемая пере-
менная». Отличие между этими понятиями заключается в том, что объявляемая
переменная может состоять из нескольких символьных лексем. Таким образом вы
можете создавать переменные с модифицирующимися именами. Если вам этого не
надо, то используйте vardef.

Средства поддержки макросов в MetaPost исключительно мощные и разнообраз-
ные. В частности, с помощью, например, инструкции primarydef можно доопреде-
лить недостающие бинарные операторы.

3.5. Стандартные функции

Лучший способ облегчить себе жизнь при написании программы, это не писать
её, а воспользоваться уже готовыми компонентами. META является специализирован-
ным языком, поэтому число стандартных функций не очень велико, но их выбор
весьма показателен.

«Из точек A и B в море вышли два корабля. . . » Требуется изобразить поверх-
ность воды: При кодировании этого рисунка использовалась функция генерации
случайных чисел:

uniformdeviate n

~v ~u

A B

α

Рис. 3.9. «Водная гладь»

В результате выполнения функции получалось случайное число в интервале [0, 1].
Кроме упомянутой функции в META есть ещё один генератор случайных чисел
normaldeviate— генерит числа по распределению Гаусса (f(x) ∼ e−x/2).

Хотелось бы упомянуть о возможности разлагать сложные объекты на составля-
ющие, например:

numeric x [ ] , y [ ] ;
pair A,B; A=(x1 , y 1 ) ;A=(x2 , y 2 ) ;
%A=(xpart x1,ypart y1)

color c ; c=(r , g , b ) ;
%c=(redpart c,greenpart c,bluepart c)

path p ;
p=A−−B;
%A = point 0 of p = point 2 of p

%B = point 1 of p = point length p of p

Таким образом можно «разобрать» на части любой путь, причём номер точки не
обязательно должен быть целым (берётся точка на линии соединения в соответствии
с дробной частью). С помощью функции length можно узнать число заданных
точек в пути, а с помощью arclength— его длину.

К уже известным вычислительным функциям sqrt, abs, mod, round, sind и cosd

полезно добавить mlog (x = 256 lnx) и mexp (x = ex/256).
Для операций с точками будут полезны функции angle (x,y)— вычисления угла

наклона к оси абсцисс для вектора ((0,0)–(x,y)) в градуcах (операция обратная
dirα) и unitvector (x,y)— единичный вектор из начала координат.

Полный список стандартных функций представлен в A User’s Manual for MetaPostД-
жона Хобби. Этот текст идёт со стандартной поставкой LATEX в виде файл mpman.pdf.

Глава 4
Графики и диаграммы

Даже жизнь можно отобразить
в виде графика функции от времени.

Отображение данных на бумаге всегда было и останется не тривиальным процес-
сом. Не смотря на то, что получение данных, как правило, занимает гораздо больше
ресурсов, задача оформления графиков достойна автоматизации. Автоматизация,
в частности позволяет, поняв как можно представить данные наилучшим образом,
в дальнейшем не снижать уровень оформления.

По хорошему графики следует создавать в специализированных приложениях.
Автоматизация этого процесса достаточно специфична, чтобы реализовывать его с
помощью программ не предназначенных только для этого. Если Вам нужны дву-
мерные графики, то никто не справится с этим лучше чем gnuplot. Если вас инте-
ресуют гистограммы, то это работа для пакетов анализа paw/cernlib или root. Но,
если Вам хочется разукрасить график, то MetaPost будет для этого очень кстати.

F , 103Н

t,с
0

0.4

0.8

1.2

0.1 0.2 0.3

Рис. 4.1. График, созданный в ручную

Ничего не мешает нарисовать график самостоятельно. Всё что для этого надо:

• знать диапазон функции и её аргумента,

• нарисовать оси координат,

• сделать решётку и поставить метки на осях,

• изобразить функцию.

В этом смысле кодирование графика ничем не отличается от кодирования рисун-
ка — всё что надо уметь это рисовать стрелки и линии. Однако, это годится только
для простых графиков.

4.1. Графики в MetaPost

Для серьёзной работы с графиками в MetaPost есть довольно продвинутый пакет
graph.mp. Этот пакет написан «отцом» MetaPost Джоном Хобби и к нему прилага-
ется подробная документация, которую можно найти в стандартной поставке LATEX
в виде файла mpgraph.pdf. Если Вы планируете воспользоваться этим пакетом, то
прежде изучите этот текст.

Рис. 4.2. Иллюстрация для объяснения понятия дифференцирования

% Файл graphics.mp

input graph ;
% График переменной скорости - тема "дифференцирование"

beginfig ( 2 ) ;
numeric u ;
u = 0 .6mm;
% начало графика

draw begingraph (12cm, 3cm) ;
path r ;
numeric e , pi ,A, sigma , n , s ca l e ,mean , y [ ] , x [ ] ;
e :=2 .718 ; p i :=3 .14159 ;A=50;

sigma =1.4 ;n :=100; s c a l e =10;mean=5;
% создание функции - она использовалась не один раз

for j :=0 upto n :
i f ( j =0):

x [ 0 ]=0 ; y [ 0 ]=0 ;
else :

x [ j ] := s c a l e ∗ j /n ;
y [ j ] := y [ j−1]+A∗( e ∗∗(−((( s c a l e ∗ j /n−mean)/
(1 . 41∗ sigma ) ) ∗ ∗2 ) ) ) ∗
(x [ j ]−x [ j −1])/( sqrt (2∗ p i )∗ sigma ) ;

f i
endfor ;
% здесь нужна функция из отрезков , поэтому step 6

r :=(0 ,0 ) for j :=0 step 6 until (n−1):
−−((2 ,0)+(x [ j ] , y [ j ] ) ) endfor . . (12 ,50)−− (15 ,50) ;

% отрисовка функции

gdraw r withpen pencircle scaled 0 . 8u ;
% сетка

autogr id ( g r id .bot , g r i d . l f t ) withcolor . 5white ;
% дополнительные метки на оси абсцисс

o t i c k . bot (btex 2\ ,\ text{с} etex , 2 ) ;
o t i c k . bot (btex 12\ ,\ text{с} etex , 1 2 ) ;
% текст и стрелка за пределами графика

gdrawdblarrow (2 ,−20)−−(12 ,−20)
withpen pencircle scaled 0 . 5u ;

g l a b e l . top (btex \(T=10\ ,\text{с }\) etex , ( 7 . 5 , −20 ) ) ;
gdrawdblarrow (13.5 ,0)−−(13 .5 ,50)

withpen pencircle scaled 0 . 5u ;
% текстовые метки внутри графика

g l a b e l . l f t (btex \(S=50\ ,\text{м}\) etex
rotated 90 , 1 / 2 [ ( 1 3 . 5 , 0 ) , ( 1 3 . 5 , 5 0 ) ] ) ;

g l a b e l . rt (btex \(v ( t )=\ f r a c {dx ( t )}{ dt }\) etex
scaled 1 . 3 , ( 0 , 2 5 ) ) ;

g l a b e l (btex \(dx\) etex , ( 1 1 . 1 , 2 0 ) ) ;
g l a b e l (btex \( dt \) etex , ( 1 0 , 9 ) ) ;
% подписи к осям

g l a b e l . rt (btex Время ( с ) etex , (13 ,−20)) ;
g l a b e l . l f t (btex Расстояние (м) etex

rotated 90 ,OUT) shifted (0cm, 0 . 5cm) ;
endgraph ;

endfig ;

В graph.mp определено специальное окружение. Между begingraph и endgraph

действует своя система координат. Эта система координат привязана ни к геомет-

рическим размерам картинки, а к диапазону осей графика. То есть точка (0,0)
соответствует точке пересечения оси абсцисс и оси ординат графика. Размер созда-
ваемого графика указывается сразу после begingraph.

Вместо стандартных команд draw, fill, label и dotlabel используются gdraw,
gfill, glabel и gdotlabel, соответственно. Новые команды работают с учётом
координат графика. С помощью них можно построить функцию именно по точкам,
не заботясь о сдвигах и тому подобное. Для подписи осей существует специальная
точка OUT, которая в зависимости от суффикса команды glabel выносит текст за
пределы основной сетки.

Функция отрисовывается точно так же как любая из кривых: создаётся путь и
выводится с помощью команды gdraw.

Для фиксации диапазона графика используется команда:

s e t r ange ( "нижний␣левый␣угол" , "верхний␣правый␣угол" ) ;

При вызове этой инструкции все последующие функции работают в указанном диа-
пазоне. Таким образом, используя одно и то же пространство можно совместить
несколько графиков, имеющих различный диапазон аргументов и функций. Шка-
лу графика можно сделать логарифмической с помощью инструкции setcoord:

% x- линейная шкала , y - логарифмическая

s e t coo rd ( l i n ea r , l o g ) ;

Для создания решётки и меток используется команда autogrid

% решётка по нижней и левой осям

autogr id ( g r id .bot , g r i d . l f t ) withcolor . 5white ;
% внешние метки по нижней оси и внутренние метки по правой оси

autogr id ( o t i c k . bot , i t i c k . rt ) ;

4.2. Работа с файлами данных

С помощью команды gdraw можно читать данные из файла. Если в качестве
аргумента передаётся текстовая строка, то gdraw предполагает, что это имя файла.

% Файл graphics.mp

% Число постов на LOR от месяца года

beginfig ( 4 ) ;
u := 0 .4mm;
draw begingraph (150u ,100u ) ;

pickup pencircle scaled 0 . 2u ;
gdraw ( " lo r −1998. dat" ) p l o t btex \(\ c i r c \) etex ;
gdraw ( " lo r −1999. dat" ) p l o t btex \(\ c i r c \) etex ;
gdraw ( " lo r −2000. dat" ) ;
gdraw ( " lo r −2001. dat" ) ;

Рис. 4.3. Число постов на LOR

gdraw ( " lo r −2002. dat" ) ;
gdraw ( " lo r −2003. dat" ) p l o t btex \(\ bu l l e t \) etex ;
gdraw ( " lo r −2004. dat" ) ;
gdraw ( " lo r −2005. dat" ) ;
gdraw ( " lo r −2006. dat" ) ;
gdraw (0 ,100)−−(12 ,100) withpen pencircle scaled 1u
dashed evenly scaled 1u withcolor blue ;
g l a b e l . top (btex 2000−−2002 etex , ( 9 , 1 0 0 ) ) withcolor blue ;
gdraw (0 ,200)−−(12 ,200) withpen pencircle scaled 1u
dashed evenly scaled 1u withcolor red ;
g l a b e l . top (btex 2004−−2006 etex , ( 9 , 2 0 0 ) ) withcolor red ;
g l a b e l . l f t (btex число <<постов>> etex rotated 90 ,OUT) ;
g l a b e l . bot (btex номер месяца в году etex ,OUT) ;

endgraph ;
endfig ;

gdraw так же как и команда draw «понимает» инструкции withpen (какое перо
использовать), withcolor (цвет линии) и dashed (использовать пунктир). Допол-
нительно gdraw воспринимает команду plot {picture}. Указанная в инструкции
plot картинка выбирается в качестве маркера. На рисунке чёрными маркерами от-
мечен 2003 год — год, когда произошло удвоение новостных постов в месяц на LOR
(http://www.linux.org.ru), а белыми маркерами— первый год существования это-
го ресурса.

По умолчанию предполагается, что входной файл представляет из себя два столб-
ца цифр, разделённые пробелами. Пример файла lor-2006.dat:

1 175

Рис. 4.4. Дефект массы химических элементов от порядкового номера в периодиче-
ской таблице Меделеева. Приведены элементы от водорода до железа.

2 181
3 184

Данные взяты с сайта LOR.
Если данные имеют более сложное представление, чем в упомянутом выше файле,

то используется команда gdata.

gdata ( "имя␣файла" , переменная куда считываются данные , действие ) ;

Если на текстовый файл вида:

1 7 . 289 H
2 2 . 425 He
−−вырезано−−
26 −57 . 710 Mn
27 −60 . 604 Fe

воздействовать с помощью следующего кода:

% Файл graphics.mp

% дефект масс

beginfig ( 5 ) ;
u := 0 .8mm;
draw begingraph (150u ,100u ) ;

gdraw "mendeleev . dat" dashed evenly scaled 1u
withcolor 0 . 5white ;

gdata ( "mendeleev . dat" , s ,
g l a b e l ( s 3 , ( scantokens s 1 , scantokens s 2 ) ) ;

Рис. 4.5. Распределение по баллам, полученных на Открытой олимпиаде по физике
в НГУ в 2004 году.

)
g l a b e l . l f t (btex Дефект массы (МэВ) etex rotated 90 ,OUT) ;
g l a b e l . bot (btex Порядковый номер элемента в таблице etex ,OUT) ;

endgraph ;
endfig ;

то получится простенькая картинка (см. рис. 4.4).
Переменная s объявляется массивом в который считывается строка. В качестве

разделителя выступает пробел. Запись s1 соответствует s[1]—первому элементу
массива. Макрос scantokens «встраивает» значение аргумента в код. В данном
случае происходит перевод строки в число. В качестве «действия» в команде gdata
может быть набор из нескольких команд.

4.3. Гистограммы в MetaPost

Если Вы хотите нарисовать гистограмму, то можно воспользоваться следующими
макросами:

% Файл graphics.mp

% делает профиль гистограммы из пути

def h i s tpa th (expr pa i r s ) =
for i=0 upto length pa i r s :

i f ( i >0):−−else : f i ( ( point i of pa i r s )−
( ( xpart (point 1 of pa i r s )−

xpart (point 0 of pa i r s ))/2 ,0))−−
( ( point i of pa i r s )+

( ( xpart (point 1 of pa i r s )−
xpart (point 0 of pa i r s ) ) / 2 , 0 ) )

endfor
enddef ;

% делает зацикленный профиль гистограммы для закрашивания

def h i s t pa th cyc l e (expr pa i r s ) =
( ( xpart (point 0 of pa i r s ) ,0)−

( ( xpart (point 1 of pa i r s )−
xpart (point 0 of pa i r s ) ) / 2 , 0 ) )

−−h i s tpa th ( pa i r s )−−
( ( xpart (point in f in i ty of pa i r s ) ,0)+

( ( xpart (point 1 of pa i r s )−
xpart (point 0 of pa i r s ))/2 ,0))−−cycle

enddef ;

Распределение представленное на картинке (рис. 4.5) похожее на убывающую
экспоненту. Вот так рисовались «пятёрки»:

% Файл graphics.mp

% Оценки

path f i v e ;
f i v e :=(21 ,2)−−(22 ,5)−−(23 ,1)−−(24 ,0)−−(25 ,1)−−

(26 ,1)−−(27 ,1)−−(28 ,3)−−(29 ,0)−−(30 ,0)−−(31 ,0);
% Заполняем гистограмму

g f i l l h i s t pa th cyc l e ( f i v e ) withcolor 0 . 6white ;

Для закрашивания контура применяется инструкция gfill— аналог команды fill.
Данные по оценкам получались в результате обработки текстовых файлов со ста-

тистикой с помощью простого скрипта на perl. Время, которое потребовалось на
автоматизацию составило примерно один человеко-день, что многократно уступает
времени написания полного отчёта куда вошли подобные гистограммы и оформле-
нию всех условий и решений Открытой олимпиады и последующих за ней летних
вступительных экзаменов.

Создание своих макросов на MetaPost является стандартным действием. META не
страдает избыточностью, поощряя «затачивать» окружения под свои нужды.

4.4. Круговые диаграммы

Пакет piechartmp.mp относительно «молодой» пакет. Замечательная, можно ска-
зать красочная, документация поставляется в виде файла piechartmp.pdf и при-
меров.

Открытая олимпиада НГУ-2005
распределение оценок ФФ-51

21 %34 %

23 % 23 %

дватри

четыре пять
Рис. 4.6. Распределение оценок

Как обычно, за русским языком надо приглядывать. По умолчанию пакет рабо-
тает только с латиницей.

% Файл pie.mp

input piechartmp ;
% Круговые гистограммы

% Открытая Олимпиада "ФФ-51"

beginfig ( 1 ) ;
numeric u ; u:=1mm;
% чтобы был русский

SetupText (1 , "\ input { preheader−base }" , "\beg in {document}" )
label (btex Открытая олимпиада НГУ−2005 etex , ( 5 u , 25u ) ) ;
label (btex распределение оценок ФФ−51 etex , ( 5 u , 20u ) ) ;
% Чтобы отображался процент

SetupPercent ( th i s , "␣\%" ) ;
% Опеределение сегментов

Segment ( 15 , "\ smal l ␣два" , auto ) ;
Segment ( 24 , "три" , auto ) ;
Segment ( 16 , "\ l a r g e ␣четыре" , auto ) ;
Segment ( 16 , "\huge␣пять" , red ) ;
% Создание круговой гистограммы

PieChart (30u , 0 . 2 , 65 , 0 , 0 ) ;
% метки

Label ( 0 ) ( percent ) ( inwards , 0 ) withcolor white ;
Label . auto ( 0 ) ( name ) ( outwards , 0 ) ;

endfig ;

Алгоритм создания круговой или секторной диаграммы следующий:

• Если Вы хотите воспользоваться автоматической системой размещения меток,
то с помощью команды SetupText следует настроить ввод текста. В файле
preheader-base.tex должна быть минимальная шапка для документа LATEX.
Если Вы хотите использовать русский язык, то там обязательно должны быть
строки вида \usepackage[T2A]{fontenc} и \usepackage[koi8-r]{inputenc}.
Вместо koi8-r можно поставить свою кодировку.

• С помощью команды Segment определить сегменты. В качестве аргументов
команде передаётся число (измеряемая величина), текстовая метка и цвет сег-
мента (можно указать значение по умолчанию auto).

Можно также ввести четвёртый необязательный строковый параметр, кото-
рый заменяет измеряемую величину при создании меток. Это решает пробле-
му ограничения на диапазон чисел в MetaPost.

• Нарисовать гистограмму. В качестве параметров команде PieChart передаёт-
ся размер, высота диаграммы, угол под которым мы на неё смотрим (трёхме-
рия), угол поворота вокруг центральной оси и «смещение» сегментов относи-
тельно центра.

• Расставить метки.

Каждому сегменту при создании присваивается порядковый номер начиная с еди-
ницы. Все изменения в круговой диаграмме делаются глобально. Пакет написан так,
чтобы было удобно работать в одном mp-файле ровно с одной диаграммой. Если при
описании диаграммы внутри окружения beginfig что-то уже определили, то нет
необходимости в последующих окружениях это повторять. Этот режим удобен если
нужно создать несколько модификаций одной диаграммы, например, для создания
«оверлеев» в презентации. Минусом такого подхода является то, что если в этом
же mp-файле хочется создать ещё одну гистограмму, то уже определённые сегменты
необходимо спрятать.

% Круговые гистограммы

% Распределение вещества во вселенной

beginfig ( 2 ) ;
numeric u ; u:=1mm;
% Прячем определённые ранее сегменты

SegmentState (1 , hidden , t h i s ) ;
SegmentState (2 , hidden , t h i s ) ;
SegmentState (3 , hidden , t h i s ) ;
SegmentState (4 , hidden , t h i s ) ;
% Опеределение новых сегментов

Segment ( 65 , "\Large␣\ bf ␣тёмная␣энергия " ,green , "\LARGE␣65\%−−70\%" ) ;
Segment ( 25 , "\Large␣\ bf ␣тёмная␣материя" , ( 1 , 0 , 1 ) , "\LARGE␣25\%" ) ;
Segment ( 0 . 5 , "звёзды" , ( 1 , 1 , 0 ) , "\LARGE␣0.5\%" ) ;

тёмная энергия

тёмная материя звёзды

обычное вещество

нейтрино

65%–70%

25%

0.5%

5%

0.3%–3%

Рис. 4.7. Распределение «вещества» в современной вселенной.

Segment ( 5 , "обычное␣вещество" , red , "\LARGE␣5\%" ) ;
Segment ( 1 , "нейтрино" , black , "\LARGE␣0.3\%−−3\%" ) ;
% Выдвигаем сегменты из диаграммы

SegmentState (7 , th i s , 0 . 7 ) ;
SegmentState (9 , th i s , 0 . 7 ) ;
SegmentState (8 , th i s , 0 . 7 ) ;
% Создаём круговой гистограммы

PieChart (30u , 0 . 3 , 30 , 340 , 0 ) ;
% Размещаем метки

Label . auto ( 0 ) ( name ) ( outwards , 0 ) ;
Label . auto ( 5 ) ( va lue ) ( inwards , 0 ) shifted (7u,−7u) withcolor white ;
Label . auto ( 6 ) ( va lue ) ( inwards , 0 ) shifted (0u , 5 u) withcolor white ;
Label . auto ( 7 ) ( va lue ) ( inwards , ( 0 u,−15u ) ) withcolor ( 1 , 1 , 0 ) ;
Label . auto ( 8 ) ( va lue ) ( inwards , ( 2 5 u ,10u ) ) withcolor red ;
Label . auto ( 9 ) ( va lue ) ( inwards , ( 2 0 u ,20u ) ) withcolor black ;

endfig ;

Состояние каждого из сегментов можно задавать с помощь команды SegmentState.

SegmentState (Порядковый номер сегмента ,
устанавливаемое состояние сегмента или t h i s если оно не меняется ,
радиальный сдвиг сегмента или t h i s если он не меняется ) ;

Состояние сегмента может быть:

• normal— сегмент становится видимым,

• invisible— сегмент не рисуется при создании круговой диаграммы (в диа-
грамме остаётся пустое место),

• hidden—при создании диаграммы этот сегмент игнорируется.

Радиальный сдвиг сегмента указывается в процентах, где 1 — это 100%, то есть
сегмент полностью «выдвинут».

Для установки меток используется команда Label:

Label (Порядковый номер сегмента ) ( метка )
( базовая точка в системе отсчёта сегмента , сдвиг ) ;

Label понимает те-же суффиксы, что и обычная команда label, плюс она понимает
суффикс auto. При установки суффикса auto Label пытается сама угадать где
лучше поставить метку.

В качестве порядкового номера сегмента можно передать 0. В этом случае ко-
манда Label применяется ко всем видимым сегментам. Label может принимать в
качестве аргумента список порядковых номеров сегментов, разделённых запятыми
(например: 5,7,9).

Метка может быть просто текстовой строкой или одним из стандартных значений:

• value—измеряемая величина, которая приписывается сегменту при его со-
здании,

• percent—процент занимаемой площади,

• name— имя сегмента, которое приписывается ему при его создании.

Базовая точка представляется в виде пары чисел (x,y), где x—расстояние от вер-
шины сегмента (0 — это вершина, 1 — это противоположный от вершины край), а y —
эквивалент полярному углу (0 — край сегмента по часовой стрелки, 1 — край сегмен-
та против часовой стрелки). В пакете определены константы inwards=(0.7,0.7) и
outwards=(1.1,0.5). Сдвиг же представлен в терминах всего графика (например,
(1cm,0)— означает сдвиг на 1 сантиметр вправо).

P.S. При желании тип заливки и порядок цветов по умолчанию можно опре-
делять самостоятельно. Подробней об этом рассказано в документации к пакету
(piechartmp.pdf).

4.5. LATEX рисует с помощью MetaPost

Для рисования графиков можно воспользоваться и возможностями самого LATEX.
Например, с этой обязанностью прекрасно справляется стандартный пакет mfpic.
Для «отрисовки» функций с помощью этого пакета достаточно задать саму функ-
цию, а не рисовать по точкам:

Рис. 4.8. График косинуса, синуса и тангенса.

% файл graphics-mfpic.tex

% Пример от Сергея В. Знаменского

\documentclass{ a r t i c l e }
\ usepackage [ ko i8−r ] { inputenc }
\usepackage [ r u s s i an ] { babel }
\ usepackage [ metapost ] { mfpic }
% Указываем mp-файл куда будет «складываться» код mfpic

\ openg r aph s f i l e {graph−mfp}
\begin {document}
% Запрет на печатать номера страницы

\ pages ty l e {empty}
% Начинаем создание графика

\mfpic [1 ] [ 57 .2 ]{−100}{300}{−1 .2}{1 .2}
% Оси координат

\ axes
\xmarks {−90 ,90 ,180 ,270}
\ymarks{−1,1}
% Рисуем синус точками

\dotted \ funct i on {−90 ,270 ,4.5}{ s ind (x )}
% Рисуем косинус пунктиром

\dashed \ funct i on {−90 ,270 ,4.5}{ cosd (x )}
% Рисуем тангенс сплошной линией

\ funct i on {−50 ,50 ,10}{ tand (x )}
\ funct i on {130 ,230 ,10}{ tand (x )}
% Заканчиваем создание графика

\ endmfpic
% Закрываем mp-файл

\ c l o s e g r a p h s f i l e
\end{document}

Для получения картинки необходимо выполнить следующие действия:

> l a t ex graph ics−mfpic . tex

> mpost graph−mfp .mp
> la t ex graph ics−mfpic . tex
> dvips −E graphics−mfpic . dvi −o graph ics−mfpic . eps

При компиляции файла graphics-mfpic.tex создаётся graph-mfp.mp, куда пишут-
ся команды на языке META. Если посмотреть на код graph-mfp.mp, то можно увидеть
что там используется макрос function, который позволяет рисовать функции одной
командой без всяких циклов. То, как это делается, можно подглядеть в библиотечке
grafbase.mp, которая входит в состав пакета mfpic.

Более подробно про пакет mfpic можно узнать из документации которую мож-
но найти в директории $(TEXMF)/texmf-dist/doc/generic/mfpic/, где $(TEXMF)—
директория в которую установлен LATEX (это заведомо верно для дистрибутива
TEXLive).

4.6. gnuplot

В случае отображения больших объёмов данных лучше использовать gnuplot.
Для того чтобы gnuplot работал с MetaPost, необходимо определить правильное
устройство вывода, то есть gnuplot следует передать команду:

set termina l mp co l o r l a t ex

0

1

2

3

4

5

6

январь 2004 май сентябрь январь 2005 май

L
in

t
(п

б
−

1
)

ВЭПП4М/КЕДР Интегральная светимость

Рис. 4.9. Интегральная светимость от времени. Данные за полтора года (примерно
130 тысяч точек) обработаны с помощью программы gnuplot.

В этом случае gnuplot будет выводить графики в формате MetaPost. К сожале-
нию, похоже, что вывод в MetaPost давно не поддерживался, поэтому установка
кодировки:

set encoding ko i8 r

не оказывает должного эффекта. Получившийся mp-файл приходится «дотачи-
вать» в ручную. Благо это не сложно (следует поправить только заголовок), но
для автоматизации следует озадачиться исправлением этой неприятной ошибки в
gnuplot.

В случае если будет желание изменить метки, то следует обратить внимание на
переменную textmag, которая отвечает за размер текста.

4.7. Подведём итоги

С помощью MetaPost можно создавать двумерные графики и диаграммы любой
сложности. Возможности языка META и стандартные пакеты позволяют сконцентри-
роваться на смысловой части и не отвлекаться на отдельные элементы оформления.

Глава 5
Дополнительные главы

Всё описать невозможно,
но ничто не запрещает попробовать.

Возможности META в то или иной мере уже изложены. В этой статье будут разо-
браны полезные приёмы, которые можно применять при кодировании картинки и
описаны некоторые из стандартных пакетов, поставляемых с MetaPost.

Исходники стандартных MetaPost пакетов, обычно находится в стандартной же
директории $(TEXMF)/texmf-dist/metapost/, а документация к ним в директории
$(TEXMF)/texmf-dist/doc/metapost/, где $(TEXMF)—корневая директория для дис-
трибутива LATEX. Сказанное верно для TEXLive.

5.1. Пакет boxes

boxes— один из самых первых пакетов общего назначения, появившихся в MetaPost.
Его предназначение — это рисовать простые диаграммы. Он проектировался как бо-
лее изощрённая замена для пакета Брайана Кернигана pic (man pic).

С другой стороны функциональности этого пакета достаточно для реализации
автоматической генерации достаточно сложных зависимостей. Для примера разбе-
рём как была реализована диаграмма объясняющая действие MetaPost-конвейер во
введении в цикл статей по MetaPost.

% Файл dop.mp

% В отличии от boxes здесь ещё определено окружение rboxit

input rboxes ;
% Определяем ещё один вид box’а

% Параметры можно передавать и так

%expr - изолированные выражения

%text - абсолютно всё, что передаётся

vardef drawshadowed (expr dx , dy ) ( text t ) =
f i x s i z e ( t ) ;

META-картинка

mpost

PostScript без шрифтов

LATEX-«обёртка»

latex+dvips

eps-файл

mptopdf

pdf-файл

Рис. 5.1. MetaPost-конвейер

forsuf f ixes s=t :
f i l l bpath . s shifted (dx , dy ) ;
unf i l l bpath . s ;
drawboxed ( s ) ;

% Можно было напечатать только текст

%и не рисовать рамку

% draw pic(s) withcolor red;

endfor ;
enddef ;

% Пример boxes

beginfig ( 2 ) ;
numeric u ; u:=1mm;
% Определяем box’ы

% Определяем box с прямыми углами

boxit . a (btex \meta−картинка etex ) ;
% Определяем box с закруглёнными углами

rbox i t . b (btex \ t e x t t t {metapost} etex ) ;
% Определяем жёсткую связь между a и b

b . n = a . s − (0 ,5u ) ;
boxit . c (btex PostSc r ip t без шрифтов etex ) ;
c . n = b . s − (0 ,5u ) ;
rbox i t . e (btex \ t e x t t t { l a t ex }\(+\)\ t e x t t t { dvips } etex ) ;
e . n=c . s−(20u ,10u ) ;
boxit . d (btex \LaTeX−<<обёртка>> etex ) ;
d . e=a .w−(5u , 0 ) ;
boxit . g (btex eps−файл etex ) ;
g . n=e . s−(0 ,5u ) ;
rbox i t . h (btex \ t e x t t t {mptopdf} etex ) ;
h . n=c . s−(−20u ,10u ) ;

boxit . i (btex pdf−файл etex ) ;
i . n=h . s−(0 ,5u ) ;
% Разрешаем зависимости и рисуем boxes

drawshadowed (1/3u,−1/3u , a , b , c , d , e , g , h , i ) ;
pickup pencircle scaled 0 . 3u ;
% Рисуем стрелки (сдвижка вида (0u,-1/3u) появилась ,

%так как было определено новое окружение drawshadowed)

drawarrow a . s −− b . n−(0u,−1/3u ) ;
drawarrow b . s −− c . n−(0u,−1/3u ) ;
drawarrow c . s {dir −90} . . {dir −90}e . n−(0u,−1/3u ) ;
drawarrow c . s {dir −90} . . {dir −90}h . n−(0u,−1/3u ) ;
drawarrow e . s −− g . n−(0u,−1/3u ) ;
drawarrow h . s −− i . n−(0u,−1/3u ) ;
drawarrow d . s {dir −90} . . {dir 0}e .w−(1/3u , 0 u ) ;

endfig ;

Обратите внимание на ещё один вид цикла forsuffixes. Эта запись позволяет
разбирать произвольное число передаваемых аргументов.

Алгоритм создания диаграммы следующий:

• С помощью команд boxit, rboxit или circleit объявляем box. Имя box’a до-
бавляется к команде через точку как суффикс. В данном примере определено
8 box’ов от «a» до «i» включительно.

• Составляем уравнения связей. Через точку к именам box’ов можно добавить
один из восьми суффиксов n— север (верхняя точка), s—юг (нижняя точ-
ка), w— запад (крайняя точка рамки слева), e— восток (крайняя точка рамки
справа) и sw, bw, se, sw—комбинации уже перечисленных суффиксов, которые
соответствуют углам рамки. Уравнение связи b.n = a.s - (0,5u); можно
описать следующим образом: северная (верхняя) точка box’a «b» находится
ниже южной (нижней) точки box’а «a» на 5u. При составления уравнения
связей следует пользоваться только знаком равенства. Это не присваивание,
а именно уравнение, которое требуется разрешить.

• С помощью команды drawboxed нарисовать «закодированную» диаграмму.
При исполнении этой команды разрешается система уравнений. По умолча-
нию, если не указать специально, box располагается в точке (0,0).

• Нарисовать стрелки между элементами диаграммы. Можно пользоваться име-
нами box’ов со стандартными суффиксами.

Прежде чем использовать этот пакет необходимо прочитать соответствующий
раздел в руководстве пользователя по MetaPost Джона Хобби mpman.pdf.

5.2. Фейнмановские диаграммы

Простота MetaPost позволяет использовать как базу для построений более высо-
кого порядка. Существует несколько подобных LATEX-пакетов. О пакете mfpic уже
упоминалось. В этом разделе описывается ещё один старейший пакет созданный по
этой технологии.

В 1995 году Торстен Охл (Torsten Ohl) представил пакет feynmp для рисования
фейнмановских диаграмм. Фейнмановские диаграммы используются для вычисле-
ния сумм большого числа вкладов от элементарных процессов. В своё время эта
технология довольно сильно продвинула технику вычислений в физике высоких
энергий. Её можно использовать везде, где сложный процесс можно описать с по-
мощью элементарных приближений.

Пакет feynmp, это пакет LATEX, который для рисования диаграмм использует
MetaPost.

Интересующий меня с целью извлечения ΓJ/ψ→e+e− процесс имеет в Борновском
приближении следующий вид:

J/ψ

e−

e+

e−

e+

Γe+e− Bre+e−

Рис. 5.2. Простейшая фейнмановская диаграмма

Это не сложная диаграмма и если не считать метки, то для её описания требуется
всего пять операторов:

% Файл eepsiee.tex

% Пакет для рисования фейнмановских диаграмм

\usepackage {feynmp}
. . .
\ beg in { f m f f i l e }{ee−psi−ee } % Имя mp-файл

\beg in { fmfgraph ∗}(110 ,62) % Размер диаграммы

\ fm f l e f t { e i , p i } % Вершины-источники

\ fmf r i gh t {eo , po} % Исходящие вершины

\ fm f l ab e l {$e^−$}{ e i } % Метка источника e^-

\ fm f l ab e l {$e^+$}{pi } % Метка источника e^+

\ fm f l ab e l {$e^+$}{po} % Метка исходящей вершины

\ fm f l ab e l {$e^−$}{eo} % Метка исходящей вершины

% Линия соединяющая источники

\ fmf{ fermion }{ e i , Ji , p i }
% Линия соединяющие исходящее вершины

\ fmf{ fermion }{po , Jo , eo}
% Метка для начальной вершины промежуточной частицы

\ fm f l ab e l {$\Gamma_{\ee }$}{ J i }
% Метка для конечной вершины промежуточной частицы

\ fm f l ab e l {$Br_{\ee }$}{Jo}
% Соединительная линия

\ fmf{heavy , l a b e l=$J/\psi $}{ Ji , Jo}
\end{ fmfgraph ∗}

\end{ f m f f i l e }

Окружение \beginfmffile в качестве параметра требует имя mp-файла в ко-
торый будут писать команды META. В данном примере имя файла определено как
ee-psi-ee.mp. Для того чтобы получить диаграмму, описанную в файле eepsiee.tex
были проделаны следующие действия:

> l a t ex e ep s i e e . tex
> mpost ee−psi−ee .mp
> la t ex e ep s i e e . tex

После выполнения этих команд результат можно посмотреть с помощью программы
xdvi или преобразовать dvi-файл в PostScript или pdf.

Следующая простейшая диаграмма не имеет особого смысла. Она просто демон-
стрирует, возможности пакета: «Сотрудничество» c MetaPost даёт возможность

J/ψ

e−

e+

q+

q−

Γe+e− Brhadrons

Рис. 5.3. Раскрашенная фейнмановская диаграмма

feynmp получить доступ к цвету. Цвет можно определять точно так же, как он
определяется в MetaPost.

% Файл eepsihadr.tex

\beg in { f m f f i l e }{ee−psi−hadr}

\beg in { fmfgraph ∗}(110 ,62)
\ fm f l e f t { i 1 , i 2}
\ fmf r i gh t {o 1 , o2}
\ fm f l ab e l {$e^−$}{ i 1}
\ fm f l ab e l {$e^+$}{ i 2}
\ fm f l ab e l {$q^+$}{o1}
\ fm f l ab e l {$q^−$}{o2}
\ fmf{ fermion , fo reground=green }{ i 1 , v1}
\ fmf{ fermion , fo reground=red }{v1 , i 2}
\ fmf{ fermion , fo reground=green }{o 1 , v 2 , v3 , v 4 , v5}
\ fmf{ fermion , fo reground=red }{v5 , v6 , v7 , v 8 , o2}
% Изменяем натяжение (расталкиваем вершины)

\ fmf{heavy , l a b e l=$J/\psi $ , t en s i on =1/3 ,
fo reground=red+green }{v1 , v5}

\ fm f f r e e z e
\ fmf{gluon , fo reground=blue }{v2 , v8}
\ fmf{gluon , fo reground=blue }{v3 , v7}
\ fmf{gluon , fo reground=blue }{v4 , v6}
\ fmfv { l a b e l =$\Gamma_{\ee }$}{v1}
\ fmfv { l a b e l=$Br_{\hadr }$ , l a b e l . d i s t =0.3w}{v5}

\end{ fmfgraph ∗}
\end{ f m f f i l e }

При создании диаграммы MetaPost пытается оптимально расположить вершины,
минимизируя взвешенную сумму расстояний между ними:

L(v1, . . . , vn) =
1

2

∑

i,j

ti,j(vi, vj)
2,

где параметр ti,j —«натяжении» между вершинами vi и vj . По умолчанию натяже-
ние равно 1. С помощью опции tension натяжение можно изменить. Чем меньше
натяжение, тем сильнее расталкиваются вершины.

Обычно, полученная диаграмма со значениями по умолчанию на требует вмеша-
тельства в код. В случае если результат не устраивает, как правило, достаточно
изменить натяжение для одного, максимум для двух соединений.

Довольно подробную документацию по пакету feynmp можно найти в файле
$(TEXMF)/texmf-dist/doc/latex/feynmf/manual.ps.gz, где $(TEXMF)—корень де-
рева LATEX (заведомо верно для TEXLive).

5.3. Фракталы

MetaPost, естественно, поддерживает рекурсию. Пользуясь этим, а так же ге-
нераторами случайных uniformdeviate или normaldeviate можно организовать
фрактальную «лесопосадку».

Рис. 5.4. Фрактальный «лес»

% Фрактальная лесопосадка

beginfig ( 1 ) ;
u:=1mm; b ranchrotat ion := 50 ;
o f f s e t := 180−branchrotat ion ;
th inn ing := 0 . 7 ;
sho r t en ing := 0 . 8 ;
def drawit (expr p , l i n e t h i c kn e s s , c o l ) =

draw p withpen pencircle scaled l i n e t h i c k n e s s withcolor co l ;
enddef ;
% A - основание B - направление роста , n - число бранчей,

%size - толщина дерева , col - цвет

vardef t r e e (expr A,B, n , s i z e , c o l ) =
save C,D, th i ckne s s ; pair C,D;
th i ckne s s := s i z e ;
C := sho r t en ing [B, A rotatedaround (B,

o f f s e t+uniformdeviate ( branchrotat ion ) ) ] ;
D := sho r t en ing [B, A rotatedaround (B,

−o f f s e t−uniformdeviate ( branchrotat ion ) ) ] ;
i f n>0:

drawit (A−−B, th i cknes s , c o l ) ;
t h i c kne s s := th inn ing ∗ th i ckne s s ;
t r e e (B, C, n−1, th i cknes s , c o l ) ;
t r e e (B, D, n−1, th i cknes s , c o l ) ;

else :
drawit (A−−B, th i cknes s , c o l ) ;
t h i c kne s s := th inn ing ∗ th i ckne s s ;
drawit (B−−C, th i cknes s , c o l ) ;
drawit (B−−D, th i cknes s , c o l ) ;

f i ;

enddef ;

numeric nbr , nx , ny , e l l , s i z e ;
color co l ;
nx :=10; ny :=5;
pair A;
for i x :=1 upto nx :

for i y :=1 upto ny :
nbr:=4+uniformdeviate 5 ;
e l l :=nbr∗u ;
x:= ix ∗(1+1/20∗normaldeviate ) ;
y:= iy ∗(1+1/20∗normaldeviate ) ;
A:=(20u∗(x+y∗ sqrt ( 2 )/2 ) , 20 u∗y∗ sqrt ( 2 ) / 2 ) ;
s i z e := e l l /5 ;
c o l :=(uniformdeviate 1 ,uniformdeviate 1 ,uniformdeviate 1 ) ;
show ix , iy ,A, e l l , nbr , s i z e , c o l ;
t r e e (A, A+(0 , e l l ) , nbr , s i z e , c o l ) ;

endfor ;
endfor ;

endfig ;

Увлечение автоматически создаваемыми картинами влечёт за собой опасность пе-
реполнения памяти. MetaPost создавался в то время, когда к используемой памяти
относились исключительно бережно.

Если при компиляции картинки будет выдана ошибка MetaPost capicity exceded,
то необходимо поправить файл конфигурации texmf.cnf. Обычно, этот файл мож-
но найти а директории $(TEXMF)/web2c/texnf.cnf1 .

За выделяемый объём памяти дляMetaPost отвечает переменная main_memory.mpost.
Значение этой переменной должно быть меньше значения main_memory, которая
ставит верхнюю границу по использованию памяти для всех TEX-подобных про-
грамм. Обратите внимание на то, что в конфигурационном файле может быть
несколько переменных с одним и тем же именем — используется последнее значение.

В моём случае ограничение по памяти стояло в 15 Мб. Очевидно, что использу-
емую память можно безболезненно увеличить. С другой стороны, если происходит
переполнение, то это возможно значит, что в алгоритм закралась ошибка.

После исправления значения переменных следует перегенерить форматные фай-
лы, например, с помощью команды texconfig init.

5.4. Увеличительное стекло

Бывает в процессе создания картинки необходимо увеличить участок для лучшей
детализации. MetaPost предоставляет средство для этого в виде команды clip,

1В TEXLive можно создавать локальные настройки

которая позволяет обрезать картинку по любому замкнутому пути. Ниже идёт код,
который позволил увеличить кусок циферблата корабельных часов.

3
~v

12

6

9 3

6

3
~u

Рис. 5.5. Преобразование Галилея

% Файл dop.mp

R:=3.6u ;
path q ;
% Определяем форму области которую хотим вырезать

q:=(−R, 0 ) . . ( R , 0 ) . . cycle ;
% Определяем центр вырезаемой области (orig)

%и местоположение увеличенного участка (copy)

pair or ig , copy ;
o r i g =(65u ,24u ) ; copy=(45u , 7 . 5 u ) ;
% Отмечаем вырезаемую область

draw q shifted o r i g dashed evenly scaled 1/2u
withpen pencircle scaled 0 . 2u wi t cho l o r red ;

picture p ;
% Сохраняем текущую картинку в переменной p

p:=currentpicture ;
% Коэффициент увеличения

numeric s c a l e ; s c a l e :=3;
% Обрезаем картинку p по замкнутому пути q

c l ip p to (q shifted o r i g ) ;
% Чистим область, где будет нарисована увеличенная копия

f i l l q scaled s c a l e shifted copy withcolor white ;
% Рисуем копию

draw p shifted −o r i g scaled s c a l e shifted copy ;
% Рисуем дополнительную стрелку и метку

%на уже увеличенной копии

draw copy withpen pencircle scaled 1u withcolor red ;

drawarrow copy−−(copy+7u∗dir −120)
withpen pencircle scaled 0 . 4u withcolor red ;

label . l r t (btex \(\ vec {u}\) etex ,
1/3 [ copy , ( copy+7u∗dir −120) ] ) ;

Пользуясь этим приёмом можно создать любой фон для какой-угодно фигуры,
сделать любую штриховку.

5.5. Штриховка

clip-технология, которая упоминалась выше, позволяет создать любой вид штри-
ховки. В стандартной поставке MetaPost идут пакета, которые этим пользуются.

Рис. 5.6. Пакет hatching.

Пакет hatching.mp представляет из себя обычный «хак»:

% Файл dop.mp

% Пример использования пакета hatching

input hatching ;
beginfig ( 3 ) ;

numeric u ; u=1mm;
path q ;
q=(10u , 0 ) . . { dir −135}(0u , 0 u){ dir 1 3 5 } . .

(−10u , 0 ) . . { dir 90}(0u , 0 u){ dir −90}.. cycle ;
h a t c h f i l l q withcolor red withcolor (25 ,1u ,−1) ;
draw q ;

endfig ;

Функция hatchfill берёт информацию о штриховке из данных, которые следу-
ют с декларацией withcolor. Сигналом, что эти данные предназначены именно
для hatchfill является то, что голубая компонента тройки чисел меньше 0. Крас-
ная компонента соответствует углу наклона штриховки, зелёная расстоянию между
штрихами. Подробности можно узнать в краткой документации к пакету.

Более развитым (возможно излишне) является пакет mpattern:

% Файл dop.mp

% Пример использования пакета mpattern

input mpattern ;

Рис. 5.7. Пакет mpattern

% Создаём "обои" ракета

beg inpattern ( ro cke t ) ;
begingroup ; save u ;

u:=1mm;
draw Rocket scaled 2/3u rotated −30;
patternbbox (−5u,−8u , 5 u , 8 u ) ;

endgroup ;
endpattern ;

% Создаём "обои" повёрнутая клетка

beg inpattern ( rotated_checker ) ;
f i l l unitsquare scaled 4mm rotated 45 withcolor . 7white ;

endpattern ;

beginfig ( 4 ) ;
numeric u ; u:=1mm;
path p ;
z1=(10u , 0 u ) ;
p=f u l l c i r c l e scaled 50u ;
% Рисуем фигуру в клетку

f i l l p withpattern rotated_checker ;
unf i l l p shifted z 1 ;
% Рисуем фигуру "в ракету"

f i l l p shifted z1 withpattern rocke t ;
endfig ;

Здесь уже определены свои команды для создания штриховки. Пакет не свободен от
недостатков, но он достаточно прост и можно легко «довести» код до необходимой
кондиции. Простота — это общее свойство пакетов MetaPost. META вынуждает писать
кратко. Подробности можно найти в документации к пакету.

5.6. Вставка eps

Пакет exteps.mp позволяет включить eps-картинку как единый объект. Краткая
документация доступна в файле exteps.pdf, поставляемом с этим пакетом.

Рис. 5.8. Добавляем чёрно-белых Туксов

% Файл dop.mp

input exteps ;
% Включение eps-картинки

beginfig ( 6 ) ;
numeric u ; u:=1mm;

% Базовая надпись

for alpha :=−90 step 3 until 0 :
label (btex LinuxFormat в России etex

scaled (5∗(1+alpha /100)) rotated alpha , ( 0 , 0 ) )
withcolor
(uniformdeviate 1 ,uniformdeviate 1 ,uniformdeviate 1 ) ;

endfor ;

% Посадим Тукса (penguin.eps) справа

beg ineps "penguin . eps " ;
% Ширина картинки

width :=30u ;
% Сдвиг картинки от начала координат

base := (60u , 5 u ) ;
% Можно нарисовать решётку на картинки

% grid := true;

% Обрезание по bounding box

% clip := true;

endeps ;

% Посадим Тукса (penguin.eps) слева

beg ineps "penguin . eps " ;
% Поворот

angle :=90;
width :=30u ;
base := (−30u,−40u ) ;
endeps ;

endfig ;

5.7. Большие числа

Когда обсуждалась вставка меток в качестве примера был представлен треуголь-
ник Паскаля. Из-за ограничений META на максимальный размер числа треугольник
Паскаля отрисовывался до 14ой строчки.

10:
1 11:

1 2 12:
1 3 3 13:

1 4 6 4 14:
1 5 10 10 5 15:

1 6 15 20 15 6 16:
1 7 21 35 35 21 7 17:

1 8 28 56 70 56 28 8 18:
1 9 36 84 126 126 84 36 9 19:

1 10 45 120 210 252 210 120 45 10 110:

1 11 55 165 330 462 462 330 165 55 11 111:

1 12 66 220 495 792 924 792 495 220 66 12 112:

1 13 78 286 715 1287 1716 1716 1287 715 286 78 13 113:

1 14 91 364 1001 2002 3003 3432 3003 2002 1001 364 91 14 114:

1 15 105 455 1365 3003 5005 6435 6435 5005 3003 1365 455 105 15 115:

1 16 120 560 1820 4368 8008 1.144×10
4

1.287×10
4

1.144×10
4 8008 4368 1820 560 120 16 116:

1 17 136 680 2380 6188 1.2376×10
4

1.9448×10
4

2.431×10
4

2.431×10
4

1.9448×10
4

1.2376×10
4 6188 2380 680 136 17 117:

1 18 153 816 3060 8568 1.8564×10
4

3.1824×10
4

4.3758×10
4

4.862×10
4

4.3758×10
4

3.1824×10
4

1.8564×10
4 8568 3060 816 153 18 118:

1 19 171 969 3876 1.1628×10
4

2.7132×10
4

5.0388×10
4

7.5582×10
4

9.2378×10
4

9.2378×10
4

7.5582×10
4

5.0388×10
4

2.7132×10
4

1.1628×10
4 3876 969 171 19 119:

1 20 190 1140 4845 1.5504×10
4

3.876×10
4

7.752×10
4

1.2597×10
5

1.6796×10
5

1.84756×10
5

Рис. 5.9. Треугольник Паскаля.

Это ограничение можно обойти воспользовавшись пакетом sarith.mp. Этот пакет
был создан специально под пакет graphics.mp. Что вполне естественно, так как
если для рисовании небольших рисунков больших чисел быть не может, то при
анализе данных это вполне рядовая ситуация.

% Файл dop.mp

% Macros for arithmetic on strings that represent big numbers

input s a r i t h ;
% Треугольник Паскаля с большими числами

beginfig ( 5 ) ;
numeric u ;

u = 1 .mm;
numeric dy , dx , x , y , i , j , sy , ds , n l a s t ; dy:=5u ;
dx:=6u ; x=0;y=0; n l a s t :=20;
% Для хранения больших чисел нужна строка

string n [ ] [ ] ;
ds =0.04 ; sy =0.032 ;
picture z ;
for i :=0 upto n l a s t :

dy:=dy∗(1−sy ) ;
y:=y−dy ;
for j :=0 upto i :

i f ( j =0) or ( j=i ) :
n [ i ] [ j ] := Sabs 1 ; % Сложение

else :
n [ i ] [ j ] :=n [ i −1] [ j −1] Sadd n [ i −1] [ j ] ; % Сложение

f i
% Формат вывода

z := thelabel ( format ( "%6g" , n [ i ] [ j ] ) , ( 0 , 0 ) ) ;
x:=dx∗( j−i / 2 ) ;
label ( z scaled (1−ds∗ i ) , ( x , y ) ) ;

endfor
z := thelabel . l f t (decimal ( i )&" : " , ( 0 , 0 ) ) ;
label ( z scaled (1−ds∗ i ) , ( dx∗(− n l a s t /2−1) ,y ) ) ;

endfor
endfig ;

Для представления больших чисел используется встроенный тип string— стро-
ка.

Операция действие

Scvnum <number> конвертация в numric
Sabs <number> абсолютное значение
<number> Sadd <number> сложение
<number> Ssub <number> вычитание
<number> Smul <number> умножение
<number> Sdiv <number> деление
<number> Sleq <number> сравнение <=

<number> Sneq <number> сравнение <>

Таблица 5.1. Операции над большими числами

Документация к пакету sarith.mp является частью документации к graphics.mp

(mpgraph.pdf от Джона Хобби).

5.8. Макрос TEX

Вы наверное уже обратили внимание, что метки в MetaPost являются статиче-
скими. Всё, что между btex и etex отдаётся LATEX для обработки. Это значит,
что метку нельзя скомпоновать из различных кусков. Решением является немного
модифицированный макрос из стандартного пакета TEX.mp:

% Файл macros.mp

vardef TEX primary s =
write " verbat imtex " to "mptextmp .mp" ;
write "\ input {preheader−base }" to "mptextmp .mp" ;
write "\beg in {document}" to "mptextmp .mp" ;
write " etex " to "mptextmp .mp" ;
write "btex ␣"&s&"␣ etex " to "mptextmp .mp" ;
write EOF to "mptextmp .mp" ;
scantokens " input ␣mptextmp"

enddef ;

В процессе вызова макроса TEX, всё что находится внутри него записывается в
mptextmp.mp, а затем этот файл включается в основной файл прямо во время компи-
ляции. То есть, происходит модификация программы прямо во время компиляции.
Пример использования макроса идёт ниже:

% Файл coord.mp

numeric u ; u:=2mm;
for i :=−15 step 5 until 15 :

label . l f t (TEX(" \( "&decimal ( i )&" \) " ) ,(−10u , i ∗u ) ) ;
endfor ;

При формировании ASCII строчки использовался оператор & для объединения. Хо-
тя, в данном случае можно было обойтись только статическими записями, но иногда
этот приём может пригодиться для целей автоматизации, где скорость компиляции
не главное.

Основной минус этого способа в том, что он очень «мееедленный». Для ускорения
следует оставить в preheader-base.tex только самые необходимые инструкции.
Что-то вроде:

\ documentclass [ 12 pt ] { a r t i c l e }
\ usepackage [ warn ] { mathtext }
\ usepackage [T2A] { fontenc }
\usepackage [ ko i8−r ] { inputenc }
\usepackage [ eng l i sh , r u s s i an ] { babel }

Глава 6
Заключение

Естественно, в таком кратком рассказе невозможно изложить всё. Основные по-
нятия и приёмы уже рассмотрены, но множество вещей выпало из обзора. В част-
ности, совсем не рассмотрены 3D объекты, создание геометрических чертежей, па-
раметризация пути и векторные поля в MetaPost. Это не смертельно, так как есть
весьма качественная свободная литература, посвящённая этим предметам.

MetaPost достаточно легко использовать после обучения, но обучение перед ис-
пользованием просто необходимо. Это свойство всех стоящих технологий, которые
позволяют делать то, для чего компьютеры и существуют: автоматизировать ру-
тинные действия, для выполнения которых не требуется задумываться. Думать же,
в любом случае, прерогатива человека.

